
� .16f84� � ;PIC 16F84に設定

W� EQU� 0� ;作業ﾚｼ゙ｽﾀ
F� EQU� 1� ;ﾌｧｲﾙ ﾚｼ゙ｽﾀ

INDIRECT�EQU� 0� ;00: 間接ｱﾄ゙ﾚｯｼﾝｸ゙
INDF� EQU� 0� ;00:
RTCC� EQU� 1� ;(RTCC = TMR0 = 1)
TMR0� EQU� 1� ;01: ﾀｲﾏ ﾃ゙ ﾀー
OPTION� EQU� 1� ;81: 任意選択制御
RBPU� EQU� 1.7� ;OPTION ﾋ゙ｯﾄ7
INTEDG� EQU� 1.6� ;OPTION ﾋ゙ｯﾄ6
RTS� EQU� 1.5� ;OPTION ﾋ゙ｯﾄ5
RTE� EQU� 1.4� ;OPTION ﾋ゙ｯﾄ4
PSA� EQU� 1.3� ;OPTION ﾋ゙ｯﾄ3
PS2� EQU� 1.2� ;OPTION ﾋ゙ｯﾄ2
PS1� EQU� 1.1� ;OPTION ﾋ゙ｯﾄ1
PS0� EQU� 1.0� ;OPTION ﾋ゙ｯﾄ0

PC� EQU� 2� ;=PCL
PCL� EQU� 2� ;02:82: ﾌ゚ﾛｸ゙ﾗﾑ ｶｳﾝﾀ下位8ﾋ゙ｯﾄ

STATUS� EQU� 3� ;03:83: ｽﾃー ﾀｽ
IRP� EQU� 3.7� ;STATUS ﾋ゙ｯﾄ7 indirect page select
RP1� EQU� 3.6� ;STATUS ﾋ゙ｯﾄ6 page select 1
RP0� EQU� 3.5� ;STATUS ﾋ゙ｯﾄ5 page select 0
TO� EQU� 3.4� ;STATUS ﾋ゙ｯﾄ4 time out bit
PD� EQU� 3.3� ;STATUS ﾋ゙ｯﾄ3 power down bit
Z� EQU� 3.2� ;STATUS ﾋ゙ｯﾄ2 zero flag
DC� EQU� 3.1� ;STATUS ﾋ゙ｯﾄ1 digit carry/borrow flag
C� EQU� 3.0� ;STATUS ﾋ゙ｯﾄ0 carry/borrow flag
c� EQU� 3.0

FSR� EQU� 4� ;04:84: 間接ｱﾄ゙ﾚｽ ﾎ゚ｲﾝﾀ
PORTA� EQU� 5� ;05: ﾎ゚ー ﾄA(RA)
RA� EQU� 5� ;05: ﾎ゚ー ﾄA(RA)
TRISA� EQU� 5� ;85: ﾎ゚ー ﾄA方向ﾚｼ゙ｽﾀ
ra� EQU� 5

PORTB� EQU� 6� ;06: ﾎ゚ー ﾄB(RB)
RB� EQU� 6� ;06: ﾎ゚ー ﾄB(RB)
TRISB� EQU� 6� ;86: ﾎ゚ー ﾄB方向ﾚｼ゙ｽﾀ
rb� EQU� 6

EEDATA� EQU� 8� ;08: EEPROMﾃ゙ ﾀー
EEADR� EQU� 9� ;09: EEPROMｱﾄ゙ﾚｽ
EECON1� EQU� 8� ;88: EEPROM制御1
EEIF� EQU� 8.4� ;88: EEPROM書き込み終了ﾌﾗｸ゙
WRERR� EQU� 8.3� ;88: EEPROM書き込み打ち切りﾌﾗｸ゙
WREN� EQU� 8.2� ;88: EEPROM書き込み許可ﾌﾗｸ゙
WR� EQU� 8.1� ;88: EEPROM書き込み開始ﾌﾗｸ゙
RD� EQU� 8.0� ;88: EEPROM読み出し開始ﾌﾗｸ゙
EECON2� EQU� 9� ;89: EEPROM制御2

PCLATH� EQU� 0AH� ;0A:8A: ﾌ゚ﾛｸ゙ﾗﾑ ｶｳﾝﾀ上位
INTCON� EQU� 0BH� ;0B:8B: 割り込み制御
GIE� EQU� 0BH.7� ;INTCON ﾋ゙ｯﾄ7
EEIE� EQU� 0BH.6� ;INTCON ﾋ゙ｯﾄ6
T0IE� EQU� 0BH.5� ;INTCON ﾋ゙ｯﾄ5
INTE� EQU� 0BH.4� ;INTCON ﾋ゙ｯﾄ4
RBIE� EQU� 0BH.3� ;INTCON ﾋ゙ｯﾄ3
T0IF� EQU� 0BH.2� ;INTCON ﾋ゙ｯﾄ2
INTF� EQU� 0BH.1� ;INTCON ﾋ゙ｯﾄ1
RBIF� EQU� 0BH.0� ;INTCON ﾋ゙ｯﾄ0

本書のﾌ゚ﾛｸ゙ﾗﾑ部分は、以下のｼﾝﾎ゙ﾙ定義済みとして記述しています。ﾌ゚ﾛｸ゙ﾗﾑ部分を利用する場合は、以下を
ｿーｽ内に記述するか、以下を定義ﾌｧｲﾙとして作成し、ｿーｽ内でｲﾝｸﾙー ﾄ゙してください。

■ はじめに

■ 内蔵ﾎ゚ ﾄー初期化

PIC内蔵のﾎ゚ ﾄーはﾋ゙ｯﾄ単位で入出力に設定可能です。通常、ﾌ゚ﾛｸ゙ﾗﾑ実行開始時に、入出力状態を決定するために
ﾎ゚ー ﾄの初期化を行います。基本的な初期化手順を次に示します。

ここでのﾎ゚ｲﾝﾄはRA,RBの方向設定以前に出力ﾃ゙ ﾀーを設定することです。この順ですと、出力ﾎ゚ー ﾄの初期値を確定す
ることが可能です。

入出力Pinの入出力決定条件は次の2点です。

① ﾘｾｯﾄ期間中は入力
② ﾘｾｯﾄ後のTRISﾚｼ゙ｽﾀ値 (1=入力,0=出力)

ﾘｾｯﾄ時のTRISﾚｼ゙ｽﾀ(Q出力)値は1(入力)で、ﾎ゚ ﾄーﾚｼ゙ｽﾀ値は不定です。
この場合、出力に設定するﾎ゚ ﾄーでは、TRISﾚｼ゙ｽﾀの該当するﾋ゙ｯﾄ=0設定
が行われるまで、対象ﾎ゚ー ﾄが入力であることに留意しなければなりません。
通常、出力Pinが接続される回路の入力は、この期間の入力値を固定する
ため、抵抗によるﾌ゚ﾙｱｯﾌ゚またはﾌ゚ﾙﾀ゙ｳﾝ処理を行います。

この処理を行っても、ﾘｾｯﾄ時に予期せぬ動作となる場合があります。その原因はﾎ゚ ﾄー ﾃ゙ ﾀーの初期化とTRISﾚｼ゙ｽﾀの
入出力設定の順序に起因します。

方向が出力に設定されると、入力(Hi-Z)から、出力ﾗｯﾁの出力をPinに出力するようになります。ﾎ゚ ﾄーﾃ゙ ﾀーの出力ﾗｯﾁ
の初期値は不定ですから、先にTRISﾚｼ゙ｽﾀで入出力を設定すると、Pin出力は不定となってしまいます。出力ﾎ゚ ﾄーの
出力ﾃ゙ ﾀー ﾗｯﾁ自身は常に有効ですので、出力として使用するﾋ゙ｯﾄには、TRISﾚｼ゙ｽﾀで方向を設定する前に、予め初
期値を設定しておきます。

通常、この後にRAMの内部変数などの初期化を行い、割り込みの初期化を行います。

■ TRIS⇒ﾎ゚ ﾄー ﾃ゙ ﾀー順での初期化

H

L

H

L
ﾎ゚ー ﾄ WR

電源

出力Pin値
H

L
Hi-Z 有効値不定値

H

L

H

L
MCLR

TRIS WR

� MOVLW� 00011100b� ;RA ﾋ゙ｯﾄ ﾃ゙ ﾀーをWに取得
� MOVWF� RA� ;RA=HHHLLに設定
� MOVLW� 00000100b� ;RB ﾋ゙ｯﾄ ﾃ゙ ﾀーをWに取得
� MOVWF� RB� ;RB=LLLLLHLLに設定
� BSF� RP0� ;ﾚｼ゙ｽﾀ ﾌｧｲﾙ1に切換
� MOVLW� 00000001b� ;RA方向ﾋ゙ｯﾄ ﾃ゙ ﾀーを設定
� MOVWF� TRISA� ;RA方向OOOOIに設定
� MOVLW� 11110001b� ;RB方向ﾋ゙ｯﾄ ﾃ゙ ﾀーを設定
� MOVWF� TRISB� ;RB方向IIIIOOOIに設定
� CLRWDT� � ;ｳｫｯﾁﾄ゙ｯｸ゙ ﾀｲﾏ初期化
� MOVLW� 00001001b� ;OPTION ﾋ゙ｯﾄ ﾃ゙ ﾀーを設定
� MOVWF� OPTION� ;ﾌ゚ﾙｱｯﾌ゚有,↓,内部,↑,WDT-4分周
� BCF� RP0� ;ﾚｼ゙ｽﾀ ﾌｧｲﾙ0に切換

D

G

Q

Q

D

G

Q

Q

Dn

ﾎ゚ー ﾄWR

ﾎ゚ー ﾄRD

TRIS WR

Pin

■ ﾎ゚ ﾄー等価回路 (1ﾋ゙ｯﾄ分)

上記例のようなﾋ゙ｯﾄ単位の検査で3ﾋ゙ｯﾄ以上となる場合は、Wﾚｼ゙ｽﾀを使用して一般の論理演算処理での分岐とする
方が命令語数、速度共に有利です。

このように、多くのCPUで記述する分岐条件を含む分岐命令の形式ではなく、分岐条件の否定を含むｽｷｯﾌ゚命令と無
条件分岐命令の組み合わせになりますので、ｽｷｯﾌ゚命令の条件記述には注意を要します。

このｽｷｯﾌ゚命令を積極的に利用すると、2つの条件のOR条件分岐が短縮して記述可能です。次は、ﾌﾗｸ゙変数FLGの
ﾋ゙ｯﾄ0か4が1の場合に、ﾗﾍ゙ﾙACTへ分岐する単純な記述です。

上記例は、次のように短縮して記述できます。

■ 条件分岐

PICには、多くのCPUに存在する条件分岐命令がありません。唯一、無条件分岐命令GOTOが存在するのみです。条
件分岐は、このGOTO命令と条件ｽｷｯﾌ゚命令を組み合わせて実現します。

中位PICには、条件ｽｷｯﾌ゚命令として次の命令があります。

� BTFSS�� f,b� ;f(ﾌｧｲﾙ:RAM)のﾋ゙ｯﾄ(b)が1で次命令をｽｷｯﾌ゚
� BTFSC�� f,b� ;f(ﾌｧｲﾙ:RAM)のﾋ゙ｯﾄ(b)が0で次命令をｽｷｯﾌ゚
� DECFSZ�� f,d� ;f(ﾌｧｲﾙ:RAM)を-1し、結果が0でｽｷｯﾌ゚
� INCFSZ�� f,d� ;f(ﾌｧｲﾙ:RAM)を+1し、結果が0でｽｷｯﾌ゚

RAM上の1ﾊ゙ｲﾄ長変数Xの値を検査し、0でﾗﾍ゙ﾙZEROへ分岐する例を次に示します。

� MOVF� X,F� ;変数Xの0検査
� BTFSC� Z� ;0以外(Z=0)でｽｷｯﾌ゚
� GOTO� ZERO� ;0(Z=1)で分岐
;

� BTFSC� FLG,0� ;FLGのﾋ゙ｯﾄ0=0でｽｷｯﾌ゚
� GOTO� ACT� ;FLGのﾋ゙ｯﾄ0=1でACTへ分岐
;
� BTFSC� FLG,4� ;FLGのﾋ゙ｯﾄ4=0でｽｷｯﾌ゚
� GOTO� ACT� ;FLGのﾋ゙ｯﾄ4=1でACTへ分岐
;

� BTFSS� FLG,0� ;FLGのﾋ゙ｯﾄ0=1でｽｷｯﾌ゚
� BTFSC� FLG,4� ;FLGのﾋ゙ｯﾄ4=0でｽｷｯﾌ゚
� GOTO� ACT� ;FLGのﾋ゙ｯﾄ4,0がともに1で分岐
;

� MOVF� FLG,W� ;FLG値をWに取得
� ANDLW� 01010101b� ;ﾋ゙ｯﾄ6,4,2,0のみ有効
� BTFSS� Z� ;ﾋ゙ｯﾄ6,4,2,0が全て0でｽｷｯﾌ゚
� GOTO� ACT� ;ﾋ゙ｯﾄ6,4,2,0の何れかが1でACTへ分岐
;

上記例は複数ﾋ゙ｯﾄのOR条件ですが、AND条件の場合は論理反転を行い、負論理AND(正論理OR)で判定します。

� MOVF� FLG,W� ;FLG値をWに取得
� XORLW� 0FFH� ;論理反転(負論理に変更)
� ANDLW� 01010101b� ;ﾋ゙ｯﾄ6,4,2,0のみ有効
� BTFSC� Z� ;ﾋ゙ｯﾄ6,4,2,0の何れかが1(元値で0)でｽｷｯﾌ゚
� GOTO� ACT� ;ﾋ゙ｯﾄ6,4,2,0のすべてが0(元値で1)でACTへ分岐
;

■ ﾃー ﾌ゙ﾙ分岐

ﾃー ﾌ゙ﾙ分岐のﾃー ﾌ゙ﾙ部分は無条件分岐命令GOTOを使用します。このﾃー ﾌ゙ﾙを参照するには、ﾃー ﾌ゙ﾙ番号をPCに加
算することで実現します。PCへの加算は、通常のﾊ゙ｲﾅﾘ加算命令ADDWFを使用します。次にﾃー ﾌ゙ﾙ分岐の基本的な
例を示します。

①行実行時のPCは次の命令位置(先頭のGOTO)を指しています。このPCにTBNを加算した値が新しいPCとなり、そ
の位置から実行されます。PICでは全ての命令が1語長命令ですので、TBNの値はﾃー ﾌ゙ﾙの番号(位置)を示すことに
なります。

■ 定数ﾃー ﾌ゙ﾙ

中位PICでは、その構造により、命令部とﾃ゙ ﾀー部が明確に分離されています。このために、多くのCPUと同様な方法で
命令部に定数を定義することができません。PICで定数を定義するには、RETLW命令を使用します。次に定数ﾃー ﾌ゙ﾙ
の例を示します。

TBN� DS� 1� ;分岐ﾃー ﾌ゙ﾙ番号変数定義

TBJ� MOVF� TBN,W� ;ﾃー ﾌ゙ﾙ番号をWに取得
� ADDWF� PC� ;ﾌ゚ﾛｸ゙ﾗﾑ ｶｳﾝﾀを対応位置に変更�� � ①
;
� GOTO� JOB0� ;ﾃー ﾌ゙ﾙ番号=0処理へ分岐
� GOTO� JOB1� ;ﾃー ﾌ゙ﾙ番号=1処理へ分岐
� GOTO� JOB2� ;ﾃー ﾌ゙ﾙ番号=2処理へ分岐

～

TBN� DS� 1� ;定数ﾃー ﾌ゙ﾙ番号変数定義

CTB� MOVF� TBN,W� ;ﾃー ﾌ゙ﾙ番号をWに取得
� ADDWF� PC� ;ﾌ゚ﾛｸ゙ﾗﾑ ｶｳﾝﾀを対応位置に変更
;
� RETLW� 100� ;W=100、呼び出し元へ復帰
� RETLW� 150� ;W=150、呼び出し元へ復帰
� RETLW� 300� ;W=300、呼び出し元へ復帰

～

基本的な動作は上記のﾃー ﾌ゙ﾙ分岐と同じです。計算されたﾃー ﾌ゙ﾙ ｱﾄ゙ﾚｽで実行されるRETLW命令が、Wﾚｼ゙ｽﾀに定
数を設定し、呼び出し元へ復帰します。このﾙー ﾁﾝを使用するには、予めTBNを設定し、CALL命令でCTBを呼び出し
ます。戻り値(Wﾚｼ゙ｽﾀ)が目的の定数になります。

■ 制限事項

上記例の方法には重要な制限事項があります。PCLATHの操作が行われていない場合、ﾃー ﾌ゙ﾙの最後が00FFH以
下でなければいけません。PCLATHが操作されている場合は、PCLATHの値が上位ｱﾄ゙ﾚｽとなる256語内にﾃー ﾌ゙ﾙが
存在していなければなりません。

これはPICの構造上の制限で、GOTOやCALL命令のように命令語内に実質的なPCH(ﾌ゚ﾛｸ゙ﾗﾑ ｶｳﾝﾀ上位)値が存在
する場合には問題となりませんが、ｵﾍ゚ﾗﾝﾄ゙に直接PCを指定する一般命令によるPC操作の場合、PCHにはPCLATH
の値が代入されることに起因します。

PC操作に関する詳細と解決法についてはPC操作を参照してください。

■ PC操作

ﾃー ﾌ゙ﾙ分岐や定数ﾃー ﾌ゙ﾙを使用する場合などに行われるPC操作では、PCH(ﾌ゚ﾛｸ゙ﾗﾑ ｶｳﾝﾀ上位)の処理に注意しな
ければいけません。これはPCの処理がGOTO命令やCALL命令の場合と異なるためです。

中位PICのGOTO,CALL命令では、その命令語内に11ﾋ゙ｯﾄのPC値を保持しています。命令実行時には、この値がPC
の下位11ﾋ゙ｯﾄになり、残りの上位ﾋ゙ｯﾄには、PCLATHの該当するﾋ゙ｯﾄ値が代入されます。

これはPCLATHのﾋ゙ｯﾄ4,3がﾌ゚ﾛｸ゙ﾗﾑ ﾒﾓﾘのﾊ゙ﾝｸ(ﾍ゚ ｼー゙)ﾚｼ゙ｽﾀのような動作となることを意味します。但し、ﾌ゚ﾛｸ゙ﾗﾑ領
域が11ﾋ゙ｯﾄ(2K語)以下場合は、この動作を考慮せずに記述することができます。

■ GOTO,CALL命令によるPC操作

511 6789 124 31012 0 511 6789 124 31012 013
PCH PCL GOTO,CALL命令

567 124 3 0

PCLATH

× × × × × ×

一般命令を使用したPC操作(書き込み)では、その命令自体で扱われるﾃ゙ ﾀーは8ﾋ゙ｯﾄですので、この8ﾋ゙ｯﾄﾃ゙ ﾀーがPCL
になり、PCHにはPCLATHの下位5ﾋ゙ｯﾄが代入されます。

■ 一般命令によるPC操作

511 6789 124 31012 0 567 124 3 0
PCH PCL ALUの値

567 124 3 0

PCLATH

× × ×

この場合、PC操作(演算)によるｷｬﾘーやﾎ゙ﾛー はPCHに反映されませんので、予めPCLATHを設定する記述が必要に
なります。次にPCLに対する加算の場合の記述例を示します。

TBN� DS� 1� ;分岐ﾃー ﾌ゙ﾙ番号変数定義

TBJ� MOVLW� TBL/256� ;ﾃー ﾌ゙ﾙ基準ｱﾄ゙ﾚｽ上位をWに取得
� MOVWF� PCLATH� ;PCLATHを仮設定
� MOVLW� TBL� ;ﾃー ﾌ゙ﾙ基準ｱﾄ゙ﾚｽ下位をWに取得�
� ADDWF� TBN,W� ;対応ﾃー ﾌ゙ﾙ ｱﾄ゙ﾚｽをWに取得
� BTFSC� C� ;ｷｬﾘーなしでｽｷｯﾌ゚
� INCF� PCLATH,F� ;PCLATCH補正(+1)
;
� MOVWF� PC� ;ﾌ゚ﾛｸ゙ﾗﾑ ｶｳﾝﾀを対応位置に変更
;
TBL� GOTO� JOB0� ;ﾃー ﾌ゙ﾙ番号=0処理へ分岐
� GOTO� JOB1� ;ﾃー ﾌ゙ﾙ番号=1処理へ分岐
� GOTO� JOB2� ;ﾃー ﾌ゙ﾙ番号=2処理へ分岐

～

この例では、ﾃー ﾌ゙ﾙの先頭ｱﾄ゙ﾚｽ上位をPCLATHに仮設定し、次にﾃー ﾌ゙ﾙ ｱﾄ゙ﾚｽ下位を加算によって計算して、その
結果のｷｬﾘーありでPCLATHを+1することで、PCへの書き込み時のPCLATHを予め設定しています。

ﾃー ﾌ゙ﾙが256境界内の場合は単純にPCLATHの設定だけの記述で済みます。また、ﾌ゚ﾛｸ゙ﾗﾑ領域が256未満(0000～
00FFH)の場合には、これらの考慮が不要です。

一般的に、ﾃー ﾌ゙ﾙ記述は256未満のｱﾄ゙ﾚｽに記述するのが最良です。

8 PIC留意点

■ ﾌ゙ﾛｯｸ転送

ﾌ゙ﾛｯｸ転送は間接ﾎ゚ｲﾝﾀを利用します。間接ﾎ゚ｲﾝﾀは1本しかありませんので、転送すべき変数領域間のｵﾌｾｯﾄ値を
加減算することで対処します。次にﾌ゙ﾛｯｸ転送ﾙー ﾁﾝの例を示します。

次に本転送ﾙー ﾁﾝの使用例を示します。この例では変数領域Xから変数領域Yへ8ﾊ゙ｲﾄの転送を行っています。

この転送ﾙー ﾁﾝでは領域の下位ｱﾄ゙ﾚｽ側から複写を行います。このため、転送元と転送先が重複し、且つ転送元が転
送先より下位側となる場合、内容が破壊されて正常な転送が行われませんので注意が必要です。

この場合は、転送ﾙー ﾁﾝの①のDECFをINCFに変更し、呼び出し時のWﾚｼ゙ｽﾀに設定する転送元ﾎ゚ｲﾝﾀを領域先頭
ではなく、領域最終位置とすることで対処できます。この変更によって領域の上位ｱﾄ゙ﾚｽ側から複写を行うようになりま
す。但し、この場合は転送元が転送先より上位側となる場合に正常な転送ができなくなります。

すべての状態で正常な転送を行うようにするには方向ﾌﾗｸ゙変数を定義し、①の位置でDECFかINCFのどちらかを使
用する条件分岐の形式に変更します。とは言え、多くの場合、領域が重複し、且つ転送元と転送先の位置関係が両
方存在する場合は殆どないでしょう。また、そのような状態にならないような変数構成などを考慮すべきです。

次に転送ﾊ゙ｲﾄ数と実行ｻｲｸﾙ数の関係を示します。

CNT� DS� 1� ;転送ﾊ゙ｲﾄ数ｶｳﾝﾀ変数定義
OFS� DS� 1� ;転送先-転送元ｵﾌｾｯﾄ変数定義
TMP� DS� 1� ;一時保存変数定義

BMV� MOVWF� FSR� ;転送元基準ｱﾄ゙ﾚｽを間接ﾎ゚ｲﾝﾀに設定
BMV1� MOVF� INDF,W� ;転送元から1ﾊ゙ｲﾄをWに取得
� MOVWF� TMP� ;転送元ﾃ゙ ﾀーを一時保存
� MOVF� OFS,W� ;転送先-転送元ｵﾌｾｯﾄをWに取得
� ADDWF� FSR,F� ;対応転送先ｱﾄ゙ﾚｽを間接ﾎ゚ｲﾝﾀに設定
� MOVF� TMP,W� ;転送元ﾃ゙ ﾀーをWに復帰
� MOVWF� INDF� ;転送元ﾃ゙ ﾀーを転送先に設定
� DECF� OFS,W� ;転送先-転送元ｵﾌｾｯﾄ-1をWに取得� � � ①
� SUBWF� FSR,F� ;転送元間接ﾎ゚ｲﾝﾀを進行/設定
� DECFSZ� CNT,F� ;転送ﾊ゙ｲﾄｶｳﾝﾀを減数,=0でｽｷｯﾌ゚
� GOTO� BMV1� ;転送ﾊ゙ｲﾄ数分継続
;
� RETURN� � ;呼出元へ復帰

X� DS� 8� ;変数領域X定義
Y� DS� 8� ;変数領域Y定義

� MOVLW� (Y-X)&0FFH�;転送先-転送元ｵﾌｾｯﾄをWに取得
� MOVWF� OFS� ;転送先-転送元ｵﾌｾｯﾄを設定
� MOVLW� Y-X� ;転送ﾊ゙ｲﾄ数をWに取得
� MOVWF� CNT� ;転送ﾊ゙ｲﾄ数を設定
� MOVLW� X� ;転送元ﾎ゚ｲﾝﾀをWに取得
� CALL� BMV� ;ﾌ゙ﾛｯｸ転送

～
～

■ 命令語数と速度の関係

転送ﾊ゙ｲﾄ数 命令語数 備考実行ｻｲｸﾙ数

実行命令ｻｲｸﾙ数=転送ﾊ゙ｲﾄ数×11 + 22 24

4 46

6 68

8
12

90

10 112

12 134

14 156

16 178

PIC留意点

■ 1の補数

1の補数は単純に各ﾋ゙ｯﾄの論理反転値です。中位PICには1の補数を作成するCOMF命令があります。8ﾋ゙ｯﾄ長の1の
補数作成は、このCOMF命令で実現できます。

多倍長の場合は単純に連続的なCOMF命令で実現します。

X� DS� 1� ;1ﾊ゙ｲﾄ長変数X定義

 � COMF� X,F� ;変数Xの1の補数作成

X� DS� 3� ;3ﾊ゙ｲﾄ長変数X定義

� COMF� X+0,F� ;変数X最下位ﾊ゙ｲﾄの1の補数作成
� COMF� X+1,F� ;変数X第2ﾊ゙ｲﾄの1の補数作成
� COMF� X+2,F� ;変数X最上位ﾊ゙ｲﾄの1の補数作成

■ 2の補数

符号付き演算処理や減算を加算命令で行う場合などに、2の補数を作成する必要が生じます。中位PICには1の補数
(ﾋ゙ｯﾄ反転)作成のCOMF命令はありますが、2の補数を作成する命令がありません。このため、2の補数を作成するに
はCOMF命令を利用して複数の命令で行います。

X� DS� 1� ;1ﾊ゙ｲﾄ長変数X定義

� COMF� X,F� ;変数Xの1の補数作成
� INCF� X,F� ;変数Xの2の補数作成

多倍長の場合は1の補数時と異なり、直前のﾊ゙ｲﾄからのｷｬﾘーを含めて処理しなければなりません。

X� DS� 2� ;2ﾊ゙ｲﾄ長変数X定義

� COMF� X+0,F� ;変数X最下位ﾊ゙ｲﾄの1の補数作成
� INCF� X+0,F� ;変数X最下位ﾊ゙ｲﾄの1の補数作成
;
� BTFSC � Z� ;最下位ﾊ゙ｲﾄからのｷｬﾘーなしでｽｷｯﾌ゚
� DECF� X+1,F� ;ｷｬﾘーで第2ﾊ゙ｲﾄ+1(実際には次補数のため-1)
� COMF� X+1,F� ;変数X第2ﾊ゙ｲﾄの2の補数作成

第2ﾊ゙ｲﾄ目以降は上記第2ﾊ゙ｲﾄ処理と同様な形式で記述します。何れの場合もｷｬﾘーをZﾌﾗｸ゙で判定(Z=1)している点
に注意してください。INCF命令ではｷｬﾘー ﾌﾗｸ゙が無効なので、直前値+1がｷｬﾘーとなる場合に必ず結果が0になること
を利用します。

■ 命令語数と速度の関係

ﾊ゙ｲﾄ数 命令語数 備考ﾋ゙ｯﾄ数 実行ｻｲｸﾙ数

8 1 2 2

16 2 5 5

24 3 8 8

32 4 11 11

40 5 14 14

48 6 17 17

56 7 20 20

64 8 23 23

10 PIC留意点

■ ﾙー ﾌ゚処理による1の補数

多倍長もﾋ゙ｯﾄ数が多くなると命令語数的にﾙー ﾌ゚処理が有利です。

X� DS� 4� ;4ﾊ゙ｲﾄ長変数X定義
CNT� DS� 1� ;作業用ｶｳﾝﾀ定義

� MOVLW� X� ;変数Xの基準ﾎ゚ｲﾝﾀをWに取得
� MOVWF� FSR� ;変数Xの基準ｱﾄ゙ﾚｽを間接ﾎ゚ｲﾝﾀに設定
� MOVLW� 4� ;変数Xのﾊ゙ｲﾄ数をWに取得
� MOVWF� CNT� ;ﾊ゙ｲﾄ ｶｳﾝﾀを初期化
;
LOOP� COMF� INDF,F� ;対応ﾊ゙ｲﾄの1の補数作成
� INCF� FSR,F� ;間接ﾎ゚ｲﾝﾀを進行(+1)
� DECFSZ� CNT,F� ;ﾊ゙ｲﾄ ｶｳﾝﾀを減数、=0でｽｷｯﾌ゚
� GOTO� LOOP� ;0以外でﾊ゙ｲﾄ数分処理継続へ

次に上記ﾙー ﾌ゚処理例と非ﾙー ﾌ゚処理(必要数分のべた書き)での命令語数と実行ｻｲｸﾙ数の関係を示します。

4364 8 8 8

■ 命令語数と速度の関係

命令語数が同一

命令語数 実行ｻｲｸﾙ数
ﾊ゙ｲﾄ数 備考ﾋ゙ｯﾄ数

非ﾙー ﾌ゚ ﾙー ﾌ゚ 非ﾙー ﾌ゚ ﾙー ﾌ゚

88 1 1 1

1316 2 2 2

1824 3 3 3

23
8

32 4 4 4

2840 5 5 5

3348 6 6 6

3856 7 7 7

上記から64ﾋ゙ｯﾄ(8ﾊ゙ｲﾄ)長を超える場合にﾙー ﾌ゚処理での命令語数の方が少なくなることが判ります。然しながら実行速
度は著しく遅くなり、同一命令語数となる64ﾋ゙ｯﾄ長の場合で5倍以上の速度低下になります。

とは言え、同一ﾌ゚ﾛｸ゙ﾗﾑ内で異なるﾋ゙ｯﾄ(ﾊ゙ｲﾄ)長のﾃ゙ ﾀーを扱う場合、上記例のFSRとCOUNTの設定だけ変更すれば
異なる長さの別ﾃ゙ ﾀーが扱えますので、命令語数的にかなり有利となります。

PIC留意点

■ ﾙー ﾌ゚処理による2の補数

多倍長もﾋ゙ｯﾄ数が多くなると命令語数的にﾙー ﾌ゚処理が有利です。

上記から32ﾋ゙ｯﾄ(4ﾊ゙ｲﾄ)長を超える場合にﾙー ﾌ゚処理での命令語数の方が少なくなることが判ります。しかし、実行速度
は著しく遅くなり、同一命令語数となる32ﾋ゙ｯﾄ長の場合で3倍以上の速度低下になります。

とは言え、同一ﾌ゚ﾛｸ゙ﾗﾑ内で異なるﾋ゙ｯﾄ(ﾊ゙ｲﾄ)長のﾃ゙ ﾀーを扱う場合、上記例のFSRとCOUNTの設定のみ変更すれば
異なる長さの別ﾃ゙ ﾀーが扱えますので、命令語数的にかなり有利となります。

X� DS� 4� ;4ﾊ゙ｲﾄ長変数X定義
CNT� DS� 1� ;作業用ｶｳﾝﾀ定義

� MOVLW� X� ;変数Xの基準ﾎ゚ｲﾝﾀをWに取得
� MOVWF� FSR� ;変数Xの基準ｱﾄ゙ﾚｽを間接ﾎ゚ｲﾝﾀに設定
� MOVLW� 4� ;変数Xのﾊ゙ｲﾄ数をWに取得
� MOVWF� CNT� ;ﾊ゙ｲﾄ ｶｳﾝﾀを初期化
� BSF� Z� ;初回用にZﾌﾗｸ゙を設定
;
LOOP� BTFSC� Z� ;前桁からのｷｬﾘーなしでｽｷｯﾌ゚
� DECF� FSR,F� ;ｷｬﾘーで対応ﾊ゙ｲﾄ+1(実際には次補数のため-1)
� COMF� FSR,F� ;対応ﾊ゙ｲﾄの2の補数作成
� INCFSZ� FSR,F� ;間接ﾎ゚ｲﾝﾀを進行(+1)
� DECFSZ� CNT� ;ﾊ゙ｲﾄ ｶｳﾝﾀを減数、=0でｽｷｯﾌ゚
� GOTO� LOOP� ;0以外でﾊ゙ｲﾄ数分処理継続へ

上記例の5行目のBSFは初回(最下位ﾊ゙ｲﾄ)処理時に7行目のDECFを実行させるために必要です。この5行目のBSF
の代わりに、GOTOで8行目のDECFへ分岐させても、結果は同じです。この場合の命令語数と実行ｻｲｸﾙ数も同じで
す。

10行目のINCFSZをINCFにしてはなりません。INCF命令はZﾌﾗｸ゙を更新します。このﾙー ﾌ゚では9行目のCOMF命令の
結果のZﾌﾗｸ゙を次回の7行目のBTFSCに引き継ぐ必要があります。INCFSZ命令はﾌﾗｸ゙を更新しません。

INCFSZ命令でFSRが0となった場合にｽｷｯﾌ゚しますが、この場合のFSRは汎用ﾚｼ゙ｽﾀ(GPR)を指していますので、多く
の場合、未実装領域となり、現実的にはほぼ有り得ません。

次に上記ﾙー ﾌ゚処理例と非ﾙー ﾌ゚処理(必要数分のべた書き)での命令語数と実行ｻｲｸﾙ数の関係を示します。

6064 8 23 23

■ 命令語数と速度の関係

命令語数が同一

命令語数 実行ｻｲｸﾙ数
ﾊ゙ｲﾄ数 備考ﾋ゙ｯﾄ数

非ﾙー ﾌ゚ ﾙー ﾌ゚ 非ﾙー ﾌ゚ ﾙー ﾌ゚

118 1 2 2

1816 2 5 5

2524 3 8 8

32
11

32 4 11 11

3940 5 14 14

4648 6 17 17

5356 7 20 20

■ ﾊ゙ｲﾅﾘ加算

中位PICには8ﾋ゙ｯﾄ(1ﾊ゙ｲﾄ)長の加算命令ADDWF命令があります。8ﾋ゙ｯﾄ(1ﾊ゙ｲﾄ)長のﾊ゙ｲﾅﾘ加算は、このADDWF命令
で実現できます。次は1ﾊ゙ｲﾄ長のX=X+Yの例です。

X� DS� 1� ;1ﾊ゙ｲﾄ長変数X定義
Y� DS� 1� ;1ﾊ゙ｲﾄ長変数Y定義

� MOVF� Y,W� ;変数Y(加算値)をWに取得
� ADDWF� X,F� ;変数Xに変数Yを加算

多倍長の場合はADDWF命令の連続で処理しますが、直前のﾊ゙ｲﾄからのｷｬﾘーを含めて処理しなければなりません。
次は2ﾊ゙ｲﾄ長のX=X+Yの例です。

X� DS� 2� ;2ﾊ゙ｲﾄ長変数X定義
Y� DS� 2� ;2ﾊ゙ｲﾄ長変数Y定義

� MOVF� Y+0,W� ;加算値最下位ﾊ゙ｲﾄをWに取得
� ADDWF � X+0,F� ;最下位ﾊ゙ｲﾄ加算
;
� MOVF� Y+1,W� ;加算値第2ﾊ゙ｲﾄをWに取得
� BTFSC� C� ;前桁からのｷｬﾘーなしでｽｷｯﾌ゚
� INCFSZ� Y+1,W� ;ｷｬﾘーで加算値+1をWに取得,=0でｽｷｯﾌ゚
� ADDWF� X+1,F� ;前桁からのｷｬﾘーなしか加算値+1≠0で第2ﾊ゙ｲﾄ加算

要点は5行目のBTFSCと6行目のINCFSZで2重の条件分岐(ｽｷｯﾌ゚)を行っていることです。前桁からのｷｬﾘーなし時に
は7行目のADDWFでの加算となり、ｷｬﾘーあり時には加算値を+1することで前桁からのｷｬﾘーを反映しています。

更に、この加算値+1が0となる場合には7行目のADDWF命令をｽｷｯﾌ゚しています。これは次桁用にｷｬﾘーﾌﾗｸ゙(=1)を保
存するためです。加算値+1が0となる場合は今桁でｷｬﾘーが発生することを意味します。7行目のADDWF命令を実行
すると、今桁の結果は正しくなりますが、ｷｬﾘーなし(C=0)になってしまいます。

この対処のために7行目のADDWF命令をｽｷｯﾌ゚する必要があります。直前のBTFSC命令の結果、INCFSZ命令が実
行される場合のｷｬﾘー ﾌﾗｸ゙は常に1です。

第2ﾊ゙ｲﾄ目以降は上記第2ﾊ゙ｲﾄ処理と同様な形式で記述します。次にﾋ゙ｯﾄ(ﾊ゙ｲﾄ)に対する命令語数と実行ｻｲｸﾙ数の
関係を示します。

■ 命令語数と速度の関係

ﾊ゙ｲﾄ数 命令語数 備考ﾋ゙ｯﾄ数 実行ｻｲｸﾙ数

8 1 2 2

16 2 6 6

24 3 10 10

32 4 14 14

40 5 18 18

48 6 22 22

56 7 26 26

64 8 30 30

PIC留意点

■ ﾙー ﾌ゚処理によるﾊ゙ｲﾅﾘ加算

多倍長もﾋ゙ｯﾄ数が多くなると命令語数的にﾙー ﾌ゚処理が有利です。

X� DS� 4� ;4ﾊ゙ｲﾄ長変数X定義
Y� DS� 4� ;4ﾊ゙ｲﾄ長変数Y定義
CNT� DS� 1� ;作業用ｶｳﾝﾀ定義
TMP� DS� 1� ;加算値一時保存変数定義
FLG� DS� 1� ;ｷｬﾘー ﾌﾗｸ゙一時保存変数定義

� MOVLW� X� ;変数Xの基準ﾎ゚ｲﾝﾀをWに取得
� MOVWF� FSR� ;変数Xの基準ｱﾄ゙ﾚｽを間接ﾎ゚ｲﾝﾀに設定
� MOVLW� 4� ;変数のﾊ゙ｲﾄ数をWに取得
� MOVWF� CNT� ;ﾊ゙ｲﾄ ｶｳﾝﾀを初期化
� BCF� C� ;初回用にｷｬﾘー ﾌﾗｸ゙をｸﾘｱ
;
LOOP� RLF� FLG,F� ;前桁からのｷｬﾘー ﾌﾗｸ゙を保存
� MOVLW� Y-X� ;変数間ｵﾌｾｯﾄをWに取得
� ADDWF� FSR,F� ;変数Yの対応ﾊ゙ｲﾄのｱﾄ゙ﾚｽを間接ﾎ゚ｲﾝﾀに設定
� MOVF� INDF,W� ;加算値をWに取得
� MOVWF� TMP� ;加算値を一時保存
� MOVLW� Y-X� ;変数間ｵﾌｾｯﾄをWに取得
� SUBWF� FSR,F� ;変数Xの対応ﾊ゙ｲﾄのｱﾄ゙ﾚｽを間接ﾎ゚ｲﾝﾀに設定
� MOVF� TMP,W� ;加算値をWに復帰
� BTFSC� FLG,0� ;前桁からのｷｬﾘーなしでｽｷｯﾌ゚
� INCFSZ� TMP,W� ;ｷｬﾘーで加算値+1をWに取得,=0でｽｷｯﾌ゚
� ADDWF� INDF,F� ;前桁からのｷｬﾘーなしか加算値+1≠0で対応ﾊ゙ｲﾄ加算
;
� INCF� FSR,F� ;間接ﾎ゚ｲﾝﾀを進行(+1)
� DECFSZ� CNT,F� ;ﾊ゙ｲﾄ ｶｳﾝﾀを減数、=0でｽｷｯﾌ゚
� GOTO� LOOP� ;0以外でﾊ゙ｲﾄ数分処理継続へ

基本的には非ﾙー ﾌ゚処理の基本形と同じ方法ですが、ﾙー ﾌ゚処理を実現するために直接ｱﾄ゙ﾚｯｼﾝｸ゙を行わず、間接ｱﾄ゙ 
ﾚｯｼﾝｸ゙によって変数ｱｸｾｽを行っています。

間接ｱﾄ゙ﾚｯｼﾝｸ゙用のﾎ゚ｲﾝﾀが1つだけですので、変数間ｵﾌｾｯﾄの加減算で変数のﾎ゚ｲﾝﾀを切り替えています。このた
め、ｻﾌ゙ﾙー ﾁﾝ化して複数変数間で使用する場合は、対象となる変数間ｵﾌｾｯﾄが同一でなければなりません。異なる
場合は8行目と12行目のMOVLW Y-XをMOVF OFS,Wのように変数OFSから対応する変数間ｵﾌｾｯﾄを取得するよう
にし、呼び出しに先立ってこの変数OFSを設定します。

次に上記ﾙー ﾌ゚処理例と非ﾙー ﾌ゚処理(必要数分のべた書き)での命令語数と実行ｻｲｸﾙ数の関係を示します。

12464 8 30 30

■ 命令語数と速度の関係

命令語数がほぼ同一

命令語数 実行ｻｲｸﾙ数
ﾊ゙ｲﾄ数 備考ﾋ゙ｯﾄ数

非ﾙー ﾌ゚ ﾙー ﾌ゚ 非ﾙー ﾌ゚ ﾙー ﾌ゚

198 1 2 2

3416 2 6 6

4924 3 10 10

64
19

32 4 14 14

7940 5 18 18

9448 6 22 22

10956 7 26 26

以降、ﾙー ﾌ゚処理の方が命令語数が少ない

上記から40ﾋ゙ｯﾄ(5ﾊ゙ｲﾄ)長を超える場合にﾙー ﾌ゚処理での命令語数の方が少なくなることが判ります。けれども実行速
度は著しく遅くなり、ほぼ同一命令語数となる40ﾋ゙ｯﾄ長の場合で4倍以上の速度低下になります。

14 PIC留意点

■ ﾊ゙ｲﾅﾘ減算

中位PICには8ﾋ゙ｯﾄ(1ﾊ゙ｲﾄ)長の減算命令SUBWF命令があります。8ﾋ゙ｯﾄ(1ﾊ゙ｲﾄ)長のﾊ゙ｲﾅﾘ減算は、このSUBWF命令
で実現できます。次は1ﾊ゙ｲﾄ長のX=X-Yの例です。

X� DS� 1� ;1ﾊ゙ｲﾄ長変数X定義
Y� DS� 1� ;1ﾊ゙ｲﾄ長変数Y定義

� MOVF� Y,W� ;変数Y(減算値)をWに取得
� SUBWF� X,F� ;変数Xから変数Yを減算

多倍長の場合はSUBWF命令の連続で処理しますが、直前のﾊ゙ｲﾄからのﾎ゙ﾛー を含めて処理しなければなりません。
次は2ﾊ゙ｲﾄ長のX=X+Yの例です。

X� DS� 2� ;2ﾊ゙ｲﾄ長変数X定義
Y� DS� 2� ;2ﾊ゙ｲﾄ長変数Y定義

� MOVF� Y+0,W� ;減算値最下位ﾊ゙ｲﾄをWに取得
� SUBWF � X+0,F� ;最下位ﾊ゙ｲﾄ減算
;
� MOVF� Y+1,W� ;減算値第2ﾊ゙ｲﾄをWに取得
� BTFSS � C� ;前桁からのﾎ゙ﾛーなしでｽｷｯﾌ゚
� INCFSZ� Y+1,W� ;ﾎ゙ﾛー で減算値+1をWに取得,=0でｽｷｯﾌ゚
� SUBWF� X+1,F� ;前桁からのﾎ゙ﾛーなしか減算値+1≠0で第2ﾊ゙ｲﾄ減算

減算での重要な点はﾎ゙ﾛー  ﾌﾗｸ゙です。ﾎ゙ﾛー  ﾌﾗｸ゙は物理位置的にｷｬﾘー ﾌﾗｸ゙と同一ですが、論理が異なります。この
ﾌﾗｸ゙をﾎ゙ﾛーとして扱う場合、負論理(0=ﾎ゙ﾛーあり、1=ﾎ゙ﾛーなし)になります。

上例の要点は5行目のBTFSSと6行目のINCFSZで2重の条件分岐(ｽｷｯﾌ゚)を行っていることです。前桁からのﾎ゙ﾛーなし
時には7行目のSUBWFでの減算となり、ﾎ゙ﾛーあり時には減算値を+1することで前桁からのﾎ゙ﾛーを反映しています。

更に、この減算値+1が0となる場合には7行目のSUBWF命令をｽｷｯﾌ゚しています。これは、次桁用にﾎ゙ﾛー  ﾌﾗｸ゙(Cﾌﾗｸ゙ 
=0)を保存するためです。減算値+1が0となる場合には今桁でﾎ゙ﾛー が発生することを意味します。7行目のSUBWF命
令を実行すると、今桁の結果は正しくなりますが、ﾎ゙ﾛーなし(Cﾌﾗｸ゙=1)になってしまいます。

この対処のため、7行目のSUBWF命令をｽｷｯﾌ゚する必要があります。直前のBTFSS命令の結果、INCFSZ命令が実行
される場合のﾎ゙ﾛー ﾌﾗｸ゙は常に(Cﾌﾗｸ゙=0)です。

第2ﾊ゙ｲﾄ目以降は上記第2ﾊ゙ｲﾄ処理と同様な形式で記述します。次にﾋ゙ｯﾄ(ﾊ゙ｲﾄ)に対する命令語数と実行ｻｲｸﾙ数の
関係を示します。

■ 命令語数と速度の関係

ﾊ゙ｲﾄ数 命令語数 備考ﾋ゙ｯﾄ数 実行ｻｲｸﾙ数

8 1 2 2

16 2 6 6

24 3 10 10

32 4 14 14

40 5 18 18

48 6 22 22

56 7 26 26

64 8 30 30

PIC留意点

■ ﾙー ﾌ゚処理によるﾊ゙ｲﾅﾘ減算

多倍長もﾋ゙ｯﾄ数が多くなると命令語数的にﾙー ﾌ゚処理が有利です。

X� DS� 4� ;4ﾊ゙ｲﾄ長変数X定義
Y� DS� 4� ;4ﾊ゙ｲﾄ長変数Y定義
CNT� DS� 1� ;作業用ｶｳﾝﾀ定義
TMP� DS� 1� ;加算値一時保存変数定義
FLG� DS� 1� ;ｷｬﾘー ﾌﾗｸ゙一時保存変数定義

� MOVLW� X� ;変数Xの基準ﾎ゚ｲﾝﾀをWに取得
� MOVWF� FSR� ;変数Xの基準ｱﾄ゙ﾚｽを間接ﾎ゚ｲﾝﾀに設定
� MOVLW� 4� ;変数のﾊ゙ｲﾄ数をWに取得
� MOVWF� CNT� ;ﾊ゙ｲﾄ ｶｳﾝﾀを初期化
� BSF� C� ;初回用にﾎ゙ﾛー ﾌﾗｸ゙をｸﾘｱ
;
LOOP� RLF� FLG,F� ;前桁からのﾎ゙ﾛー ﾌﾗｸ゙を保存
� MOVLW� Y-X� ;変数間ｵﾌｾｯﾄをWに取得
� ADDWF� FSR,F� ;変数Yの対応ﾊ゙ｲﾄのｱﾄ゙ﾚｽを間接ﾎ゚ｲﾝﾀに設定
� MOVF� INDF,W� ;減算値をWに取得
� MOVWF� TMP� ;減算値を一時保存
� MOVLW� Y-X� ;変数間ｵﾌｾｯﾄをWに取得
� SUBWF� FSR,F� ;変数Xの対応ﾊ゙ｲﾄのｱﾄ゙ﾚｽを間接ﾎ゚ｲﾝﾀに設定
� MOVF� TMP,W� ;減算値をWに復帰
� BTFSS� FLG,0� ;前桁からのﾎ゙ﾛーなしでｽｷｯﾌ゚
� INCFSZ� TMP,W� ;ﾎ゙ﾛー で減算値+1をWに取得,=0でｽｷｯﾌ゚
� SUBWF� INDF,F� ;前桁からのﾎ゙ﾛーなしか減算値+1≠0で対応ﾊ゙ｲﾄ減算
;
� INCF� FSR,F� ;間接ﾎ゚ｲﾝﾀを進行(+1)
� DECFSZ� CNT,F� ;ﾊ゙ｲﾄ ｶｳﾝﾀを減数、=0でｽｷｯﾌ゚
� GOTO� LOOP� ;0以外でﾊ゙ｲﾄ数分処理継続へ

基本的には非ﾙー ﾌ゚処理の基本形と同じ方法ですが、ﾙー ﾌ゚処理を実現するために直接ｱﾄ゙ﾚｯｼﾝｸ゙を行わず、間接ｱﾄ゙ 
ﾚｯｼﾝｸ゙によって変数ｱｸｾｽを行います。

間接ｱﾄ゙ﾚｯｼﾝｸ゙用のﾎ゚ｲﾝﾀが1つだけですので、変数間ｵﾌｾｯﾄの加減算で変数のﾎ゚ｲﾝﾀを切り替えます。このため、
ｻﾌ゙ﾙー ﾁﾝ化して複数変数間で使用する場合は、対象となる変数間ｵﾌｾｯﾄが同じでなければなりません。異なる場合
は8行目と12行目のMOVLW Y-XをMOVF OFS,Wのように変数OFSから対応する変数間ｵﾌｾｯﾄを取得するようにし、
呼び出しに先立ってこの変数OFSを設定します。

次に上記ﾙー ﾌ゚処理例と非ﾙー ﾌ゚処理(必要数分のべた書き)での命令語数と実行ｻｲｸﾙ数の関係を示します。

12464 8 30 30

■ 命令語数と速度の関係

命令語数がほぼ同一

命令語数 実行ｻｲｸﾙ数
ﾊ゙ｲﾄ数 備考ﾋ゙ｯﾄ数

非ﾙー ﾌ゚ ﾙー ﾌ゚ 非ﾙー ﾌ゚ ﾙー ﾌ゚

198 1 2 2

3416 2 6 6

4924 3 10 10

64
19

32 4 14 14

7940 5 18 18

9448 6 22 22

10956 7 26 26

以降、ﾙー ﾌ゚処理の方が命令語数が少ない

上記から40ﾋ゙ｯﾄ(5ﾊ゙ｲﾄ)長を超える場合にﾙー ﾌ゚処理での命令語数の方が少なくなることが判ります。けれども実行速
度は著しく遅くなり、ほぼ同一命令語数となる40ﾋ゙ｯﾄ長の場合で4倍以上の速度低下になります。

16 PIC留意点

■ 加算によるﾊ゙ｲﾅﾘ乗算

中位PICには乗算命令がありませんので、加算命令ADDWFを基に作成します。基本的な方法は通常の筆算による
乗算と同一です。大きく異なるのは通常の10進数の場合、乗数の各桁計算時に非乗数×乗数の対応桁(1桁)の乗算
がありますが、2進数の場合の1桁は0か1だけですので、0でそのまま、1で非乗数の加算になり、加算のみで行えるこ
とです。

下図は8ﾋ゙ｯﾄ(1ﾊ゙ｲﾄ)長のX×Y=Xの例です。結果のXは16ﾋ゙ｯﾄ(2ﾊ゙ｲﾄ)長になります。例では中間結果を保持する作業
領域をXの上位に連結し、中間結果とXのﾋ゙ｯﾄ ｼﾌﾄを同時に行っています。この手法により、作業領域が最小となり、
最終結果の移動もなくなります。また、各桁判定が常に変数Xの最下位ﾋ゙ｯﾄとなり、その結果の桁単位の加算位置も
固定になりますので、ﾙー ﾌ゚処理に最適な方法です。

変数Xの上位は初期値として0を設定しておく必要があります。

151413121110 9 8 7 6 5 4 3 2 1 0

0 0 0 0 0 0 0 0 0 0 0 0 01 1

1 0 0 0 001 1
7 6 5 4 3 2 1 0

15 8 7 0

0 0 0 0 0 0 0 0 0 0 0 00 1 1

1 0 0 0 001 1
+

15 8 7 0

0 0 0 0 0 0 0 0 0 1

07

0 1 1 1

15 8 7 0

0 0 00 0 0 0 0 0 01

10

01 1 1

C

15 8 7 0

0 0 00 0 0 0 00 101 1 1

1

1

1

1

10

0

15 8 7 0

0 0 0 00 0 0 01

C

1

15 8 7 0

0 0 0 0 00 101 1 1 1011 1

01 1 11 1

15 8 7 0

0 0 0 00 0 0 01 10 1 1 11

11

15 8 7 0

0 0 0 00 0 0 010 1 1 11 11

15 8 7 0

0 0 0 00 0 0 010 1 1 11 11

15 8 7 0

0 0 0 000 0 010 1 1 11 11

15 8 7 0

0 0 00 00 0 010 1 1 11 11

1 0 0 0 001 1
+ 07

1 0 0 0 001 1
+ 07

0

0

0

0

0

Xの最下位ﾋ゙ｯﾄ=0なので結果とXを1ﾋ゙ｯﾄ右ｼﾌﾄし、
最上位ﾋ゙ｯﾄ=0とします。

Xのﾋ゙ｯﾄ1(現在の最下位ﾋ゙ｯﾄ)=1なので、結果の対
応桁とYを加算し、結果とXを1ﾋ゙ｯﾄ右ｼﾌﾄします。最
上位ﾋ゙ｯﾄは加算結果のｷｬﾘー(例では0)です。

Xのﾋ゙ｯﾄ2(現在の最下位ﾋ゙ｯﾄ)=0なので、結果とXを
1ﾋ゙ｯﾄ右ｼﾌﾄし、最上位ﾋ゙ｯﾄ=0とします。

Xのﾋ゙ｯﾄ3(現在の最下位ﾋ゙ｯﾄ)=1なので、結果の対
応桁とYを加算し、結果とXを1ﾋ゙ｯﾄ右ｼﾌﾄします。最
上位ﾋ゙ｯﾄは加算結果のｷｬﾘー(例では0)です。

Xのﾋ゙ｯﾄ4(現在の最下位ﾋ゙ｯﾄ)=1なので、結果の対
応桁とYを加算し、結果とXを1ﾋ゙ｯﾄ右ｼﾌﾄします。最
上位ﾋ゙ｯﾄは加算結果のｷｬﾘー(例では1)です。

Xのﾋ゙ｯﾄ5(現在の最下位ﾋ゙ｯﾄ)=0なので、結果とXを
1ﾋ゙ｯﾄ右ｼﾌﾄし、最上位ﾋ゙ｯﾄ=0とします。

Xのﾋ゙ｯﾄ6(現在の最下位ﾋ゙ｯﾄ)=0なので、結果とXを
1ﾋ゙ｯﾄ右ｼﾌﾄし、最上位ﾋ゙ｯﾄ=0とします。

Xの最上位ﾋ゙ｯﾄ(現在の最下位ﾋ゙ｯﾄ)=0なので、結果
とXを1ﾋ゙ｯﾄ右ｼﾌﾄし、最上位ﾋ゙ｯﾄ=0とします。

X

Y

この方法は任意のﾊ゙ｲﾄ長乗算が可能です。次に任意のﾊ゙ｲﾄ長乗算作成時の要点を示します。

① 変数Xの上位に連続して同一長の領域を確保します。
② 上記領域は0で初期化します。
③ 加算はﾊ゙ｲﾄ長分の多倍長加算で行います。
④ 右ｼﾌﾄはﾊ゙ｲﾄ長分のRRF命令の連続で行います。

PIC留意点

■ ﾊ゙ｲﾅﾘ乗算

中位PICには乗算命令がありませんので、加算命令ADDWFを基に作成します。次は16ﾋ゙ｯﾄ(2ﾊ゙ｲﾄ)長のX×Y=X(32 
ﾋ゙ｯﾄ)の例です。

X� DS� 2+2� ;2+2ﾊ゙ｲﾄ長変数X定義
Y� DS� 2� ;2ﾊ゙ｲﾄ長変数Y定義
CNT� DS� 1� ;作業用ﾋ゙ｯﾄ ｶｳﾝﾀ

� CLRF� X+2+0� ;結果上位を初期化
� CLRF� X+2+1� ;
;
� MOVLW� 16� ;ﾋ゙ｯﾄ数をWに取得
� MOVWF� CNT� ;ﾋ゙ｯﾄ ｶｳﾝﾀを初期化
LOOP� BCF� C� ;対応桁ｷｬﾘーなしを仮設定(ｷｬﾘー ﾌﾗｸ゙をｸﾘｱ)
� BTFSS� X+0,0� ;変数Xの対応ﾋ゙ｯﾄ=1でｽｷｯﾌ゚
� GOTO� LP1� ;変数Xの対応ﾋ゙ｯﾄ=0で分岐
;
� MOVF� Y+0,W� ;乗数Y最下位ﾊ゙ｲﾄをWに取得
� ADDWF� X+2+0,F� ;対応桁結果最下位ﾊ゙ｲﾄ計算
;
� MOVF� Y+1,W� ;乗数Y第2ﾊ゙ｲﾄをWに取得
� BTFSC� C� ;前桁からのｷｬﾘーなしでｽｷｯﾌ゚
� INCFSZ� Y+1,W� ;ｷｬﾘーで乗数第2ﾊ゙ｲﾄ値+1をWに取得,=0でｽｷｯﾌ゚
� ADDWF� X+2+1,F� ;前桁からのｷｬﾘーなしか第2ﾊ゙ｲﾄ値+1≠0で第2ﾊ゙ｲﾄ計算
;
LP1� RRF� X+2+1,F� ;対応桁ｷｬﾘー,結果,元ﾃ゙ ﾀーX右ｼﾌﾄ
� RRF� X+2+0,F
� RRF� X+1,F� ;
� RRF� X+0,F� ;
;
� DECFSZ� CNT,F� ;ﾋ゙ｯﾄ ｶｳﾝﾀ減数,=0でｽｷｯﾌ゚
� GOTO� LOOP� ;≠0で、ﾋ゙ｯﾄ数分処理継続へ

3ﾊ゙ｲﾄ長以上の場合は1,2行目の結果上位初期化部と18～21行目の右ｼﾌﾄ部への追加、4行目のﾋ゙ｯﾄ数の変更、多
倍長加算の追加を行います。多倍長加算は17行目からへ13～16行目の第2ﾊ゙ｲﾄ加算と同様の形式で記述します。
ﾗﾍ゙ﾙLP1からのRRF命令は必ず上位側から記述しなければなりません。

次にﾋ゙ｯﾄ(ﾊ゙ｲﾄ)長に対する命令語数と実行ｻｲｸﾙ数の関係を示します。

■ 命令語数と速度の関係

ﾊ゙ｲﾄ数 命令語数 備考ﾋ゙ｯﾄ数 実行ｻｲｸﾙ数

8 1 12 74～82

16 2 19 179～259

24 3 26 316～532

32 4 33 485～901

40 5 40 686～1366

48 6 47 919～1927

56 7 54 1184～2584

64 8 61 1481～3337

8×8 = 16ﾋ゙ｯﾄ

16×16 = 32ﾋ゙ｯﾄ

24×24 = 48ﾋ゙ｯﾄ

32×32 = 64ﾋ゙ｯﾄ

40×40 = 80ﾋ゙ｯﾄ

48×48 = 96ﾋ゙ｯﾄ

56×56 = 112ﾋ゙ｯﾄ

64×64 = 128ﾋ゙ｯﾄ

18 PIC留意点

■ 兼用型ﾊ゙ｲﾅﾘ乗算

基本形を変形した兼用型ﾊ゙ｲﾅﾘ乗算ﾙー ﾁﾝの例を次に示します。この例では8,16,24,32ﾋ゙ｯﾄ長の乗算が行えます。但
し、各々の結果は8×8がX+3から、16×16がX+2から、24×24がX+1から、32×32がX+0からになります。

X� DS� 4+4� ;4+4ﾊ゙ｲﾄ長変数X定義
Y� DS� 4� ;4ﾊ゙ｲﾄ長変数Y定義
CNT� DS� 1� ;作業用ﾋ゙ｯﾄ ｶｳﾝﾀ

MUL8� MOVLW� 8� ;ﾋ゙ｯﾄ数(=8)をWに取得� [8×8入口]
� GOTO� MUL� ;実処理へ
;
MUL16� MOVLW� 16� ;ﾋ゙ｯﾄ数(=16)をWに取得� [16×16入口]
� GOTO� MUL� ;実処理へ
;
MUL24� MOVLW� 24� ;ﾋ゙ｯﾄ数(=24)をWに取得� [24×24入口]
� GOTO� MUL� ;実処理へ
;
MUL32� MOVLW� 32� ;ﾋ゙ｯﾄ数(=32)をWに取得� [32×32入口]
;
MUL� CLRF� X+4+0� ;結果上位を初期化
� CLRF� X+4+1� ;
� CLRF� X+4+2� ;
� CLRF� X+4+3� ;
;
� MOVWF� CNT� ;ﾋ゙ｯﾄ ｶｳﾝﾀを初期化
MUL1� BCF� C� ;対応桁ｷｬﾘーなしを仮設定(ｷｬﾘー ﾌﾗｸ゙をｸﾘｱ)
� BTFSS� X+0,0� ;変数Xの対応ﾋ゙ｯﾄ=1でｽｷｯﾌ゚
� GOTO� MUL2� ;変数Xの対応ﾋ゙ｯﾄ=0で分岐
;
� MOVF� Y+0,W� ;乗数Y最下位ﾊ゙ｲﾄをWに取得
� ADDWF� X+4+0,F� ;対応桁結果最下位ﾊ゙ｲﾄ計算
;
� MOVF� Y+1,W� ;乗数Y第2ﾊ゙ｲﾄをWに取得
� BTFSC� C� ;前桁からのｷｬﾘーなしでｽｷｯﾌ゚
� INCFSZ� Y+1,W� ;ｷｬﾘーで乗数第2ﾊ゙ｲﾄ値+1をWに取得,=0でｽｷｯﾌ゚
� ADDWF� X+4+1,F� ;前桁からのｷｬﾘーなしか第2ﾊ゙ｲﾄ値+1≠0で第2ﾊ゙ｲﾄ計算
;
� MOVF� Y+2,W� ;乗数Y第3ﾊ゙ｲﾄをWに取得
� BTFSC� C� ;前桁からのｷｬﾘーなしでｽｷｯﾌ゚
� INCFSZ� Y+2,W� ;ｷｬﾘーで乗数第3ﾊ゙ｲﾄ値+1をWに取得,=0でｽｷｯﾌ゚
� ADDWF� X+4+2,F� ;前桁からのｷｬﾘーなしか第3ﾊ゙ｲﾄ値+1≠0で第3ﾊ゙ｲﾄ計算
;
� MOVF� Y+3,W� ;乗数Y第4ﾊ゙ｲﾄをWに取得
� BTFSC� C� ;前桁からのｷｬﾘーなしでｽｷｯﾌ゚
� INCFSZ� Y+3,W� ;ｷｬﾘーで乗数第4ﾊ゙ｲﾄ値+1をWに取得,=0でｽｷｯﾌ゚
� ADDWF� X+4+3,F� ;前桁からのｷｬﾘーなしか第4ﾊ゙ｲﾄ値+1≠0で第4ﾊ゙ｲﾄ計算
;
MUL2� RRF� X+4+3,F� ;対応桁ｷｬﾘー,結果,元ﾃ゙ ﾀーX右ｼﾌﾄ
� RRF� X+4+2,F� ;
� RRF� X+4+1,F� ;
� RRF� X+4+0,F� ;
� RRF� X+3,F� ;
� RRF� X+2,F� ;
� RRF� X+1,F� ;
� RRF� X+0,F� ;
;
� DECFSZ� CNT,F� ;ﾋ゙ｯﾄ ｶｳﾝﾀ減数,=0でｽｷｯﾌ゚
� GOTO� MUL1� ;≠0で、ﾋ゙ｯﾄ数分処理継続へ
;
� RETURN� � ;呼び出し元へ復帰

■ 減算によるﾊ゙ｲﾅﾘ除算

中位PICには除算命令がありませんので減算命令SUBWFを基に作成します。基本的な方法は通常の筆算による除
算と同じです。大きく異なるのは通常の10進数の場合、除数の各桁計算時に非除数の対応桁÷除数の除算がありま
すが、2進数の場合の1桁は0か1だけですので、0でそのまま、1で非除数の減算になり、減算のみで行えることです。

下図は8ﾋ゙ｯﾄ(1ﾊ゙ｲﾄ)長のX÷Y=Xの例です。商は8ﾋ゙ｯﾄ(1ﾊ゙ｲﾄ)長でX+0に、剰余も8ﾋ゙ｯﾄ(1ﾊ゙ｲﾄ)長でX+1に格納されま
す。例では中間結果を保持する作業領域をXの上位に連結し、中間結果とXのﾋ゙ｯﾄ ｼﾌﾄを同時に行っています。この
手法によって作業領域が最小となり、最終結果の移動もなくなります。また、各桁での減算位置が常に固定となり、そ
の結果の商ﾋ゙ｯﾄも常に変数Xの最下位ﾋ゙ｯﾄとなりますので、ﾙー ﾌ゚処理に最適な方法です。

変数Xの上位は初期値として0を設定しておく必要があります。

ｷｬﾘー ﾌﾗｸ゙は非除数ﾋ゙ｯﾄ数＞除数ﾋ゙ｯﾄ数の場合のみ考慮しなければなりません。その他の場合のｷｬﾘー ﾌﾗｸ゙は常に0
となりますので、剰余－除数の対象にはなりません。

左ｼﾌﾄ後、最上位桁減算不可なので、商の最上位
ﾋ゙ｯﾄ=0を(現在の商最下位ﾋ゙ｯﾄへ)設定します。

左ｼﾌﾄ後、上位2桁での減算不可なので、商の第
7ﾋ゙ｯﾄ=0を(現在の商最下位ﾋ゙ｯﾄへ)設定します。

左ｼﾌﾄ後、上位6桁での減算可なので減算を行い、
商の第3ﾋ゙ｯﾄ=1を(現在の商最下位ﾋ゙ｯﾄへ)設定しま
す。

X

Y

1

剰余 商

左ｼﾌﾄ後、上位3桁での減算不可なので、商の第
6ﾋ゙ｯﾄ=0を(現在の商最下位ﾋ゙ｯﾄへ)設定します。

左ｼﾌﾄ後、上位4桁での減算不可なので、商の第
5ﾋ゙ｯﾄ=0を(現在の商最下位ﾋ゙ｯﾄへ)設定します。

左ｼﾌﾄ後、上位5桁での減算不可なので、商の第
4ﾋ゙ｯﾄ=0を(現在の商最下位ﾋ゙ｯﾄへ)設定します。

左ｼﾌﾄ後、上位7桁での減算不可なので、商の第
2ﾋ゙ｯﾄ=0を(現在の商最下位ﾋ゙ｯﾄへ)設定します。

左ｼﾌﾄ後、上位8桁での減算可なので減算を行い、
商の最下位ﾋ゙ｯﾄ=1を設定します。

この方法は任意のﾊ゙ｲﾄ長除算が可能です。次に任意のﾊ゙ｲﾄ長除算作成時の要点を示します。

① 変数Xの上位に連続して同一長の領域を確保します。
② 上記領域は0で初期化します。
③ 減算はﾊ゙ｲﾄ長分の多倍長減算で行います。
④ 左ｼﾌﾄはﾊ゙ｲﾄ長分のRLF命令の連続で行います。

■ ﾊ゙ｲﾅﾘ除算

中位PICには除算命令がありませんので、減算命令SUBWFを基に作成します。次は16ﾋ゙ｯﾄ(2ﾊ゙ｲﾄ)長のX÷Y= X(16 
ﾋ゙ｯﾄ)の例です。

X� DS� 2+2� ;2+2ﾊ゙ｲﾄ長変数X定義
Y� DS� 2� ;2ﾊ゙ｲﾄ長変数Y定義
CNT� DS� 1� ;作業用ﾋ゙ｯﾄ ｶｳﾝﾀ

� CLRF� X+2+0� ;非除数上位を初期化
� CLRF� X+2+1� ;
;
� MOVLW� 16� ;ﾋ゙ｯﾄ数をWに取得
� MOVWF� CNT� ;ﾋ゙ｯﾄ ｶｳﾝﾀを初期化
LOOP� BCF� C� ;商の対応ﾋ゙ｯﾄ=0を仮設定
� RLF� X+0,F� ;非除数,商を左ｼﾌﾄ(桁移動)
� RLF� X+1,F� ;
� RLF� X+2+0,F� ;
� RLF� X+2+1,F� ;
;
;� BTFSC� C� ;非除数対象最上位桁=0でｽｷｯﾌ゚� �非除数ﾋ゙ｯﾄ数＞
;� GOTO� LP2� ;非除数対象最上位桁=1で減算へ� (�除数ﾋ゙ｯﾄ数時必要)
;
� MOVF� Y+1,W� ;除数第2ﾊ゙ｲﾄをWに取得
� SUBWF� X+2+1,W� ;第2ﾊ゙ｲﾄ減算可/不可検査
� BTFSS� Z� ;同一値で下位桁検査へｽｷｯﾌ゚
� GOTO� LP1� ;可/不可判定可で分岐
;
� MOVF� Y+0,W� ;除数最下位ﾊ゙ｲﾄをWに取得
� SUBWF� X+2+0,W� ;最下位ﾊ゙ｲﾄ減算可/不可検査
LP1� BTFSS� C� ;減算可でｽｷｯﾌ゚
� GOTO� LP3� ;減算不可で分岐
;
LP2� BSF� X+0,0� ;商の対応ﾋ゙ｯﾄ=1を設定
� MOVF� Y+0,W� ;除数最下位ﾊ゙ｲﾄをWに取得
� SUBWF� X+2+0,F� ;最下位ﾊ゙ｲﾄ減算
;
� MOVF� Y+1,W� ;除数第2ﾊ゙ｲﾄをWに取得
� BTFSS� C� ;前桁からのﾎ゙ﾛーなしでｽｷｯﾌ゚
� INCFSZ� Y+1,W� ;ﾎ゙ﾛー で減数値+1をWに取得,=0でｽｷｯﾌ゚
� SUBWF� X+2+1,F� ;第2ﾊ゙ｲﾄ減算
;
LP3� DECFSZ� CNT,F� ;ﾋ゙ｯﾄ ｶｳﾝﾀ減数,=0でｽｷｯﾌ゚
� GOTO� LOOP� ;≠0で、ﾋ゙ｯﾄ数分処理継続へ

3ﾊ゙ｲﾄ長以上の場合は1,2行目の非除数上位初期化部と7～10行目の左ｼﾌﾄ部への追加、4行目のﾋ゙ｯﾄ数の変更、多
倍長比較と多倍長減算の追加を行います。多倍長比較は14行目からに15～18行目の第2ﾊ゙ｲﾄ比較と同様の形式で
記述し、多倍長減算は33行目からに29～32行目の第2ﾊ゙ｲﾄ減算と同様の形式で記述します。7行目からのRLF命令
は必ず下位側から記述しなければなりません。

次にﾋ゙ｯﾄ(ﾊ゙ｲﾄ)長に対する命令語数と実行ｻｲｸﾙ数の関係を示します。

■ 命令語数と速度の関係

ﾊ゙ｲﾄ数 命令語数 備考ﾋ゙ｯﾄ数 実行ｻｲｸﾙ数

8 1 15 90～106

16 2 26 259～371

24 3 37 436～796

32 4 48 645～1381

40 5 59 886～2126

48 6 60 1159～3031

56 7 71 1464～4096

64 8 82 1801～5321

8÷8 = 8ﾋ゙ｯﾄ (剰余=8ﾋ゙ｯﾄ)

16÷16 = 16ﾋ゙ｯﾄ (剰余=16ﾋ゙ｯﾄ)

24÷24 = 24ﾋ゙ｯﾄ (剰余=24ﾋ゙ｯﾄ)

32÷32 = 32ﾋ゙ｯﾄ (剰余=32ﾋ゙ｯﾄ)

40÷40 = 40ﾋ゙ｯﾄ (剰余=40ﾋ゙ｯﾄ)

48÷48 = 48ﾋ゙ｯﾄ (剰余=48ﾋ゙ｯﾄ)

56÷56 = 56ﾋ゙ｯﾄ (剰余=56ﾋ゙ｯﾄ)

64÷64 = 64ﾋ゙ｯﾄ (剰余=64ﾋ゙ｯﾄ)

■ 兼用型ﾊ゙ｲﾅﾘ除算

基本形を変形した兼用型ﾊ゙ｲﾅﾘ除算ﾙー ﾁﾝの例を次に示します。この例では8～64ﾋ゙ｯﾄ長の除算が行えます。結果は
商がX+0から、剰余がX+8からになります。但し、非除数Xの設定値は次表の位置から設定しなければなりません。

X� DS� 8+4� ;8+4ﾊ゙ｲﾄ長変数X定義
Y� DS� 4� ;4ﾊ゙ｲﾄ長変数Y定義
CNT� DS� 1� ;作業用ﾋ゙ｯﾄ ｶｳﾝﾀ

DV6432� MOVLW� 64� ;ﾋ゙ｯﾄ数(=64)をWに取得� � [64÷32入口]
� GOTO� DIV32� ;実処理へ
;
DV5632� MOVLW� 56� ;ﾋ゙ｯﾄ数(=56)をWに取得� � [56÷32入口]
� GOTO� DIV32� ;実処理へ
;
DV4832� MOVLW� 48� ;ﾋ゙ｯﾄ数(=48)をWに取得� � [48÷32入口]
� GOTO� DIV32� ;実処理へ
;
DV4824� MOVLW� 48� ;ﾋ゙ｯﾄ数(=48)をWに取得� � [48÷24入口]
� GOTO� DIV24� ;実処理へ
;
DV4032� MOVLW� 40� ;ﾋ゙ｯﾄ数(=40)をWに取得� � [40÷32入口]
� GOTO� DIV32� ;実処理へ
;
DV4024� MOVLW� 40� ;ﾋ゙ｯﾄ数(=40)をWに取得� � [40÷24入口]
� GOTO� DIV24� ;実処理へ
;
DV3232� MOVLW� 32� ;ﾋ゙ｯﾄ数(=32)をWに取得� � [32÷32入口]
� GOTO� DIV32� ;実処理へ
;
DV3216� MOVLW� 32� ;ﾋ゙ｯﾄ数(=32)をWに取得� � [32÷16入口]
� GOTO� DIV16� ;実処理へ
;
DV2424� MOVLW� 24� ;ﾋ゙ｯﾄ数(=24)をWに取得� � [24÷24入口]
� GOTO� DIV24� ;実処理へ
;
DV2416� MOVLW� 24� ;ﾋ゙ｯﾄ数(=24)をWに取得� � [24÷16入口]
� GOTO� DIV16� ;実処理へ
;
DV1616� MOVLW� 16� ;ﾋ゙ｯﾄ数(=16)をWに取得� � [16÷16入口]
� GOTO� DIV16� ;実処理へ
;
DV1608� MOVLW� 16� ;ﾋ゙ｯﾄ数(=16)をWに取得� � [16÷8入口]
� GOTO� DIV8� ;実処理へ
;
DV0808� MOVLW� 8� ;ﾋ゙ｯﾄ数(=8)をWに取得� � [8÷8入口]
;
DIV8� CLRF� Y+1� ;除数上位を初期化� � (8ﾋ゙ｯﾄ除数移行点)
DIV16� CLRF� Y+2� ;�� � � � (16ﾋ゙ｯﾄ除数移行点)
DIV24� CLRF� Y+3� ;�� � � � (24ﾋ゙ｯﾄ除数移行点)
DIV32� CLRF� X+8+0� ;非除数(仮想)上位を初期化� (32ﾋ゙ｯﾄ除数移行点)
� CLRF� X+8+1� ;
� CLRF� X+8+2� ;
� CLRF� X+8+3� ;
� MOVWF� CNT� ;ﾋ゙ｯﾄ ｶｳﾝﾀを初期化
;

非除数Xﾋ゙ｯﾄ長 4016 24 5632 48 648

X+2 X+0X+1X+6X+7 X+3X+5 X+4非除数X設定位置

備考 8÷8
16÷8
16÷ 16

24÷16
24÷24

32÷16
32÷32

40÷24
40÷32

48÷24
48÷32

56÷32 64÷32

次頁へ続く

;
DIV1� BCF� C� ;商の対応ﾋ゙ｯﾄ=0を仮設定
� RLF� X+0,F� ;非除数,商を左ｼﾌﾄ(桁移動)
� RLF� X+1,F� ;
� RLF� X+2,F� ;
� RLF� X+3,F� ;
� RLF� X+4,F� ;
� RLF� X+5,F� ;
� RLF� X+6,F� ;
� RLF� X+7,F� ;
� RLF� X+8+0,F� ;各桁減算(剰余)部左ｼﾌﾄ(桁移動)
� RLF� X+8+1,F� ;
� RLF� X+8+2,F� ;
� RLF� X+8+3,F� ;
;
� BTFSC� C� ;非除数対象最上位桁=0でｽｷｯﾌ゚
� GOTO� DIV3� ;非除数対象最上位桁=1で減算へ
;
� MOVF� Y+3,W� ;除数第4ﾊ゙ｲﾄをWに取得
� SUBWF� X+8+3,W� ;第4ﾊ゙ｲﾄ減算可/不可検査
� BTFSS� Z� ;同一値で下位桁検査へｽｷｯﾌ゚
� GOTO� DIV2� ;可/不可判定可で分岐
;
� MOVF� Y+2,W� ;除数第3ﾊ゙ｲﾄをWに取得
� SUBWF� X+8+2,W� ;第3ﾊ゙ｲﾄ減算可/不可検査
� BTFSS� Z� ;同一値で下位桁検査へｽｷｯﾌ゚
� GOTO� DIV2� ;可/不可判定可で分岐
;
� MOVF� Y+1,W� ;除数第2ﾊ゙ｲﾄをWに取得
� SUBWF� X+8+1,W� ;第2ﾊ゙ｲﾄ減算可/不可検査
� BTFSS� Z� ;同一値で下位桁検査へｽｷｯﾌ゚
� GOTO� DIV2� ;可/不可判定可で分岐
;
� MOVF� Y+0,W� ;除数最下位ﾊ゙ｲﾄをWに取得
� SUBWF� X+8+0,W� ;最下位ﾊ゙ｲﾄ減算可/不可検査
DIV2� BTFSS� C� ;減算可でｽｷｯﾌ゚
� GOTO� DIV4� ;減算不可で分岐
;
DIV3� BSF� X+0,0� ;商の対応ﾋ゙ｯﾄ=1を設定
� MOVF� Y+0,W� ;除数第1ﾊ゙ｲﾄをWに取得
� SUBWF� X+8+0,F� ;第1ﾊ゙ｲﾄ減算
;
� MOVF� Y+1,W� ;除数第2ﾊ゙ｲﾄをWに取得
� BTFSS� C� ;前桁からのﾎ゙ﾛーなしでｽｷｯﾌ゚
� INCFSZ� Y+1,W� ;ﾎ゙ﾛー で減数値+1をWに取得,=0でｽｷｯﾌ゚
� SUBWF� X+8+1,F� ;前桁からのﾎ゙ﾛーなしか減算値+1≠0で第2ﾊ゙ｲﾄ減算
;
� MOVF� Y+2,W� ;除数第3ﾊ゙ｲﾄをWに取得
� BTFSS� C� ;前桁からのﾎ゙ﾛーなしでｽｷｯﾌ゚
� INCFSZ� Y+2,W� ;ﾎ゙ﾛー で減数値+1をWに取得,=0でｽｷｯﾌ゚
� SUBWF� X+8+2,F� ;前桁からのﾎ゙ﾛーなしか減算値+1≠0で第3ﾊ゙ｲﾄ減算
;
� MOVF� Y+3,W� ;除数第4ﾊ゙ｲﾄをWに取得
� BTFSS� C� ;前桁からのﾎ゙ﾛーなしでｽｷｯﾌ゚
� INCFSZ� Y+3,W� ;ﾎ゙ﾛー で減数値+1をWに取得,=0でｽｷｯﾌ゚
� SUBWF� X+8+3,F� ;前桁からのﾎ゙ﾛーなしか減算値+1≠0で第4ﾊ゙ｲﾄ減算
;
DIV4� DECFSZ� CNT,F� ;ﾋ゙ｯﾄ ｶｳﾝﾀ減数,=0でｽｷｯﾌ゚
� GOTO� DIV1� ;≠0で、ﾋ゙ｯﾄ数分処理継続へ
;
� RETURN� � ;呼び出し元へ復帰

■ ﾊ゙ｲﾅﾘ⇒ﾊ゚ｯｸ化BCD変換

8ﾋ゙ｯﾄ長ﾊ゙ｲﾅﾘ⇒ﾊ゚ｯｸ化BCD変換の例を以下に示します。

X� DS� 1� ;ﾊ゙ｲﾅﾘ値(元値)変数定義
Y� DS� 2� ;ﾊ゚ｯｸ化BCD値(変換値)変数定義
CNT1� DS� 1� ;ﾊ゙ｲﾅﾘ用ﾋ゙ｯﾄ ｶｳﾝﾀ定義
CNT2� DS� 1� ;BCD用ﾊ゙ｲﾄ ｶｳﾝﾀ定義

B2BCD� CLRF� Y+0� ;結果BCD変数を初期化� � � ①
� CLRF� Y+1� ;
� MOVLW� 8� ;ﾋ゙ｯﾄ ｶｳﾝﾀを設定�� � � ②
� MOVWF� CNT1� ;
� GOTO� B2D3� ;初回開始へ
;
B2D1� MOVLW� Y+0� ;BCD最下位ﾊ゙ｲﾄ ﾎ゚ｲﾝﾀをWに取得
� MOVWF� FSR� ;BCD基準ｱﾄ゙ﾚｽを間接ﾎ゚ｲﾝﾀに設定
� MOVLW� 3� ;BCDﾊ゙ｲﾄ数をWに取得� � � ③
� MOVWF� CNT2� ;BCDﾊ゙ｲﾄ ｶｳﾝﾀを設定
B2D2� MOVLW� 8-10/2� ;下桁で8-次回10以上となる値(5)を取得
� ADDWF� INDF,F� ;下桁次回10>か検査(次桁LSB,3ﾋ゙ｯﾄ補整)
� BTFSS� INDF,3� ;下桁次回10以上でｽｷｯﾌ゚
� SUBWF� INDF,F� ;次回9以下で元値復帰
;
� MOVLW� (8-10/2)*16�;上桁で8-次回10以上となる値(5)を取得
� ADDWF� INDF,F� ;上桁次回10>か検査(次桁LSB,3ﾋ゙ｯﾄ補整)
� BTFSS� INDF,7� ;上桁次回10以上でｽｷｯﾌ゚
� SUBWF� INDF,F� ;次回9以下時、元値復帰
;
� INCF� FSR,F� ;間接ﾎ゚ｲﾝﾀを進行
� DECFSZ� CNT2,F� ;ﾊ゙ｲﾄ ｶｳﾝﾀを減数,=0でｽｷｯﾌ゚
� GOTO� B2D2� ;BCDﾊ゙ｲﾄ数分継続
;
� RLF� X+0,F� ;ﾊ゙ｲﾅﾘ上位ﾋ゙ｯﾄから処理�� � ④
� RLF� Y+0,F� ;ﾊ゙ｲﾅﾘ上位ﾋ゙ｯﾄをBCD最下位ﾋ゙ｯﾄへ� ⑤
� RLF� Y+1,F� ;
� DECFSZ� CNT1,F� ;ﾋ゙ｯﾄｶｳﾝﾀを減数,=0でｽｷｯﾌ゚
� GOTO� B2D1� ;ﾋ゙ｯﾄ数分継続

上記例は変数定義と以下の点を変更することで、8ﾋ゙ｯﾄ長以外でも使用できます。

① 結果BCD値のﾊ゙ｲﾄ数分を初期化します。
② ﾋ゙ｯﾄ ｶｳﾝﾀ値はﾊ゙ｲﾅﾘ値のﾋ゙ｯﾄ数を設定します。
③ BCDﾊ゙ｲﾄ数は結果となるﾊ゚ｯｸ化BCDのﾊ゙ｲﾄ数を設定します。
④ ﾊ゙ｲﾅﾘ値のﾊ゙ｲﾄ数分、下位ﾊ゙ｲﾄからRLF命令を記述します。
⑤ 結果BCD値ﾊ゙ｲﾄ数分、下位ﾊ゙ｲﾄからRLF命令を記述します。

ﾊ゙ｲﾅﾘ値のﾋ゙ｯﾄ長とBCD桁数、ﾊ゚ｯｸ化BCDﾊ゙ｲﾄ数の関係を次に示します。

ﾊ゙ｲﾅﾘ ﾋ゙ｯﾄ長 4016 24 5632 48 648

15 201753 138 10BCD桁数

ﾊ゚ｯｸ化BCDﾊ゙ｲﾄ数 8 10932 74 5

■ 命令語数と速度の関係

ﾊ゙ｲﾄ数 命令語数 備考ﾋ゙ｯﾄ数 実行ｻｲｸﾙ数

8 1 25 240

16 2 28 717

24 3 31 1418

32 4 34 2343

40 5 39 4001

48 6 42 5478

56 7 45 7179

64 8 48 9104

0 ～ 255

0 ～ 65535

0 ～ 16777215

0 ～ 4294967295

0 ～ 1099511627775

0 ～ 281474976710655

0 ～ 72057594037927935

0 ～ 18446744073709551615

■ 兼用型ﾊ゙ｲﾅﾘ⇒ﾊ゚ｯｸ化BCD変換

基本形を変形した兼用型ﾊ゙ｲﾅﾘ⇒ﾊ゚ｯｸ化BCD変換ﾙー ﾁﾝの例を次に示します。基本形に対し、既に終了(ｼﾌﾄ)した無
効なﾊ゙ｲﾅﾘ値分の処理をﾊ゙ｲﾄ単位で行わないことで高速化を計っています。この手法で基本形に対して平均で2倍程
度速くなりますが、命令語数が増大します。また、ﾊ゙ｲﾅﾘ値によって実行速度が変化し、基本形に対して約1.5～3倍程
度の速度になります。

本例では8～64ﾋ゙ｯﾄ長のﾊ゙ｲﾅﾘ⇒ﾊ゚ｯｸ化BCD変換が行えます。結果のﾊ゚ｯｸ化BCD値はY+0からになります。元値の
ﾊ゙ｲﾅﾘ値は次表の位置から設定しなければなりません。

ﾊ゙ｲﾅﾘ値ﾋ゙ｯﾄ長 4016 24 5632 48 648

X+2 X+0X+1X+6X+7 X+3X+5 X+4ﾊ゙ｲﾅﾘ値X設定位置

15 201753 138 10BCD桁数

ﾊ゚ｯｸ化BCDﾊ゙ｲﾄ数 8 10932 74 5

X� DS� 8� ;ﾊ゙ｲﾅﾘ値(元値)変数定義
Y� DS� 10� ;ﾊ゚ｯｸ化BCD値(変換値)変数定義
CNT1� DS� 1� ;ﾊ゙ｲﾅﾘ用ﾋ゙ｯﾄ ｶｳﾝﾀ定義
CNT2� DS� 1� ;BCD用ﾊ゙ｲﾄ ｶｳﾝﾀ定義
TMP� DS� 1� ;処理BCDﾊ゙ｲﾄ数(高速化用)

B2D8� MOVLW� 8� ;ﾊ゙ｲﾅﾘ ﾋ゙ｯﾄ数(=8)をWに取得� [8ﾋ゙ｯﾄ ﾊ゙ｲﾅﾘ入口]
� GOTO� BD8� ;実処理へ
;
B2D16� MOVLW� 16� ;ﾊ゙ｲﾅﾘ ﾋ゙ｯﾄ数(=16)をWに取得� [16ﾋ゙ｯﾄ ﾊ゙ｲﾅﾘ入口]
� GOTO� BD16� ;実処理へ
;
B2D24� MOVLW� 24� ;ﾊ゙ｲﾅﾘ ﾋ゙ｯﾄ数(=24)をWに取得� [24ﾋ゙ｯﾄ ﾊ゙ｲﾅﾘ入口]
� GOTO� BD24� ;実処理へ
;
B2D32� MOVLW� 32� ;ﾊ゙ｲﾅﾘ ﾋ゙ｯﾄ数(=32)をWに取得� [32ﾋ゙ｯﾄ ﾊ゙ｲﾅﾘ入口]
� GOTO� BD32� ;実処理へ
;
B2D40� MOVLW� 40� ;ﾊ゙ｲﾅﾘ ﾋ゙ｯﾄ数(=40)をWに取得� [40ﾋ゙ｯﾄ ﾊ゙ｲﾅﾘ入口]
� GOTO� BD40� ;実処理へ
;
B2D48� MOVLW� 48� ;ﾊ゙ｲﾅﾘ ﾋ゙ｯﾄ数(=48)をWに取得� [48ﾋ゙ｯﾄ ﾊ゙ｲﾅﾘ入口]
� GOTO� BD48� ;実処理へ
;
B2D56� MOVLW� 56� ;ﾊ゙ｲﾅﾘ ﾋ゙ｯﾄ数(=56)をWに取得� [56ﾋ゙ｯﾄ ﾊ゙ｲﾅﾘ入口]
� GOTO� BD56� ;実処理へ
;
B2D64� MOVLW� 64� ;ﾊ゙ｲﾅﾘ ﾋ゙ｯﾄ数(=64)をWに取得� [64ﾋ゙ｯﾄ ﾊ゙ｲﾅﾘ入口]
;
BD64� CLRF� Y+9� ;結果BCD変数を初期化
BD56� CLRF� Y+8� ;
BD48� CLRF� Y+7� ;
BD40� CLRF� Y+6� ;
� CLRF� Y+5� ;
BD32� CLRF� Y+4� ;
BD24� CLRF� Y+3� ;
BD16� CLRF� Y+2� ;
BD8� CLRF� Y+1� ;
� CLRF� Y+0� ;
� MOVWF� CNT1� ;ﾋ゙ｯﾄ ｶｳﾝﾀを設定
� MOVLW� 1� ;Wに1を取得
� MOVWF� TMP� ;処理BCDﾊ゙ｲﾄ数を初期化
� GOTO� B2D5� ;初回開始へ
;

次頁へ続く

;
B2D1� MOVLW� Y+0� ;BCD最下位ﾊ゙ｲﾄ ﾎ゚ｲﾝﾀをWに取得
� MOVWF� FSR� ;BCD最下位ﾊ゙ｲﾄ ｱﾄ゙ﾚｽを間接ﾎ゚ｲﾝﾀに設定
� MOVF� TMP,W� ;処理BCDﾊ゙ｲﾄ数をWに取得
� MOVWF� CNT2� ;BCDﾊ゙ｲﾄ ｶｳﾝﾀを設定
B2D2� MOVLW� 8-10/2� ;下桁で8-次回10以上となる値(5)を取得
� ADDWF� INDF,F� ;下桁次回10>か検査(次桁LSB,3ﾋ゙ｯﾄ補整)
� BTFSS� INDF,3� ;下桁次回10以上でｽｷｯﾌ゚
� SUBWF� INDF,F� ;次回9以下で元値復帰
;
� MOVLW� (8-10/2)*16�;上桁で8-次回10以上となる値(5)を取得
� ADDWF� INDF,F� ;上桁次回10>か検査(次桁LSB,3ﾋ゙ｯﾄ補整)
;� BTFSS� INDF,7� ;上桁次回10以上でｽｷｯﾌ゚
;� SUBWF� INDF,F� ;次回9以下で元値復帰

� BTFSS� INDF,7� ;上桁次回10以上でｽｷｯﾌ゚
� GOTO� B2D3� ;次回9以下で分岐
;
� DECF� CNT2,W� ;最下位BCDﾊ゙ｲﾄか検査
� BTFSC� Z� ;最下位ﾊ゙ｲﾄ以外でｽｷｯﾌ゚
� INCF� TMP,F� ;最下位ﾊ゙ｲﾄで処理BCDﾊ゙ｲﾄ数を増加(+1)
� GOTO� B2D4� ;次へ

B2D3� SUBWF� INDF,F� ;次回9以下時、元値復帰
;
B2D4� INCF� FSR,F� ;間接ﾎ゚ｲﾝﾀを進行
� DECFSZ� CNT2,F� ;ﾊ゙ｲﾄ ｶｳﾝﾀを減数,=0でｽｷｯﾌ゚
� GOTO� B2D2� ;BCDﾊ゙ｲﾄ数分継続
;
B2D5� RLF� X+0,F� ;ﾊ゙ｲﾅﾘ上位ﾋ゙ｯﾄから処理
� RLF� X+1,F� ;
� RLF� X+2,F� ;
� RLF� X+3,F� ;
� RLF� X+4,F� ;
� RLF� X+5,F� ;
� RLF� X+6,F� ;
� RLF� X+7,F� ;
� RLF� Y+0,F� ;ﾊ゙ｲﾅﾘ上位ﾋ゙ｯﾄをBCD最下位ﾋ゙ｯﾄへ
� RLF� Y+1,F� ;
� RLF� Y+2,F� ;
� RLF� Y+3,F� ;
� RLF� Y+4,F� ;
� RLF� Y+5,F� ;
� RLF� Y+6,F� ;
� RLF� Y+7,F� ;
� RLF� Y+8,F� ;
� RLF� Y+9,F� ;
� DECFSZ� CNT1,F� ;ﾋ゙ｯﾄ ｶｳﾝﾀを減数,=0でｽｷｯﾌ゚
� GOTO� B2D1� ;ﾋ゙ｯﾄ数分継続へ
;
� RETURN� � ;呼出元へ復帰

■ ﾊ゚ｯｸ化BCD⇒ﾊ゙ｲﾅﾘ変換

ﾊ゚ｯｸ化BCDをﾊ゙ｲﾅﾘに変換するにはBCDの各桁のﾊ゙ｲﾅﾘ値を加算します。BCDの各桁を下位桁からd1,d2,･･･とする
と、n桁のBCD値は次式で表せます。

� d1＋d2×101＋d3×102＋・・・＋dn×10n-1

この式を変形すると次式になります。

� d1＋(d2＋(d3＋・・・dn×10・・・)×10)×10

この式からBCDの各桁毎に10倍と加算を行うことで変換できることが判ります。

次に4桁ﾊ゚ｯｸ化BCD⇒16ﾋ゙ｯﾄ長ﾊ゙ｲﾅﾘ変換の例を示します。

X� DS� 2+2� ;ﾊ゙ｲﾅﾘ値(変換値)+一時保存変数定義
Y� DS� 2� ;ﾊ゚ｯｸ化BCD値(元値)変数定義

D2B� CLRF� X+1� ;結果ﾊ゙ｲﾅﾘ変数を初期化�� ①
;
� SWAPF� Y+1,W� ;BCD最上位桁値をWに取得� ②
� ANDLW� 0FH� ;下位桁のみ有効
� MOVWF� X+0� ;BCD最上位桁値設定
� CALL� M10� ;BCD最上位桁値を10倍
� MOVF� Y+1,W� ;BCD第3桁値をWに取得
� CALL� C1D� ;BCD第3桁値を積算後10倍
� SWAPF� Y,W� ;BCD第2桁値をWに取得
� CALL� C1D� ;BCD第2桁値を積算後10倍
� MOVF� Y,W� ;最下位桁値をWに取得

;� ( BCD1桁加算 )

A1D� ANDLW� 0FH� ;下位桁のみ有効
� ADDWF� X+0,F� ;最下位ﾊ゙ｲﾄに加算
� MOVLW� 1� ;Wに1を設定
� BTFSC� C� ;ｷｬﾘーなしでｽｷｯﾌ゚�� � ③
� ADDWF� X+1,F� ;ｷｬﾘーで第2ﾊ゙ｲﾄ+1�� � ③
� RETURN� � ;呼出元へ復帰

;� ( BCD1桁を加算後10倍 )

C1D� CALL� A1D� ;BCD1桁加算
;
M10� BCF� C� ;ｷｬﾘー ﾌﾗｸ゙を解除
� RLF� X+0,F� ;最下位ﾊ゙ｲﾄ×2�� � ④
� RLF� X+1,F� ;第2ﾊ゙ｲﾄ×2
;
� MOVF� X+0,W� ;最下位ﾊ゙ｲﾄ×2値をWに取得� ⑤
� MOVWF� X+2+0� ;最下位ﾊ゙ｲﾄ×2値を一時保存
� MOVF� X+1,W� ;第2ﾊ゙ｲﾄ×2値をWに取得
� MOVWF� X+2+1� ;第2ﾊ゙ｲﾄ×2値を一時保存
;
� BCF� C� ;ｷｬﾘー ﾌﾗｸ゙を解除
� RLF� X+0,F� ;最下位ﾊ゙ｲﾄ×2(×4)� � ④
� RLF� X+1,F� ;第2ﾊ゙ｲﾄ×2(×4)
;
� BCF� C� ;ｷｬﾘー ﾌﾗｸ゙を解除
� RLF� X+0,F� ;最下位ﾊ゙ｲﾄ×2(×8)� � ④
� RLF� X+1,F� ;第2ﾊ゙ｲﾄ×2(×8)
;
� MOVF� X+2+0,W� ;加算値最下位ﾊ゙ｲﾄをWに取得
� ADDWF� X+0,F� ;最下位ﾊ゙ｲﾄ加算
;
� MOVF� X+2+1,W� ;加算値第2ﾊ゙ｲﾄをWに取得�� ⑥
� BTFSC� C� ;前桁からのｷｬﾘーなしでｽｷｯﾌ゚
� INCFSZ� X+2+1,W� ;ｷｬﾘーで加数第2ﾊ゙ｲﾄ+1をWに取得,=0でｽｷｯﾌ゚
� ADDWF� X+1,F� ;前桁or第2ﾊ゙ｲﾄ+1のｷｬﾘーなしで第2ﾊ゙ｲﾄ加算
;
� RETURN� � ;呼出元へ復帰

前記の例は、変数と以下の点を変更することで、BCD4桁以外の桁数にも使用できます。

① 結果ﾊ゙ｲﾅﾘ値ﾊ゙ｲﾄ数-1分、初期化します。
② 同様の形式で必要なBCD桁数分を記述します。
③ この2行と同様な形式で、結果ﾊ゙ｲﾅﾘ値のﾊ゙ｲﾄ数-1分、加算処理を記述します。
④ 必要なﾊ゙ｲﾅﾘ値のﾊ゙ｲﾄ数分、下位ﾊ゙ｲﾄからRLF命令を記述します。
⑤ 結果ﾊ゙ｲﾅﾘ値ﾊ゙ｲﾄ数分の一時保存処理を記述します。
⑥ 同様な形式で、結果ﾊ゙ｲﾅﾘ値ﾊ゙ｲﾄ数-1分、加算処理を記述します。

ﾊ゙ｲﾅﾘ値のﾋ゙ｯﾄ長とBCD桁数、ﾊ゚ｯｸ化BCDﾊ゙ｲﾄ数の関係を次に示します。

ﾊ゙ｲﾅﾘ ﾋ゙ｯﾄ長 4017 24 5430 47 647

ﾊ゚ｯｸﾄBCDﾊ゙ｲﾄ数 103 4 51

10

2

14 20 27 34 37

6

44

7

50

8

57

9

60

14 191652 127 9BCD桁数 3 4 6 8 10 11 13 15 17 18

ﾊ゙ｲﾅﾘ ﾊ゙ｲﾄ数 31 2 4 5 6 7 8

■ 命令語数と速度の関係

ﾊ゙ｲﾅﾘ ﾊ゙ｲﾄ数 命令語数 備考BCD桁数 実行ｻｲｸﾙ数

2 1 19 23 0 ～ 99

3 2 37 68 0 ～ 999

4 2 41 101 0 ～ 9999

5 3 57 179 0 ～ 99999

6 3 61 223 0 ～ 999999

7 3 65 267 0 ～ 9999999

8 4 81 389 0 ～ 99999999

9 4 85 444 0 ～ 999999999

10 5 101 0 ～ 9999999999599

11 5 105 0 ～ 99999999999665

12 5 109 0 ～ 999999999999731

13 6 121 0 ～ 9999999999999930

14 6 125 0 ～ 999999999999991007

15 7 141 0 ～ 9999999999999991239

16 7 145 0 ～ 99999999999999991327

17 8 161 0 ～ 999999999999999991592

18 8 165 0 ～ 9999999999999999991691

19 8 169 0 ～ 99999999999999999991790

■ 兼用型ﾊ゚ｯｸ化BCD⇒ﾊ゙ｲﾅﾘBCD変換

基本形を変形した兼用型ﾊ゚ｯｸ化BCD⇒ﾊ゙ｲﾅﾘ変換ﾙー ﾁﾝの例を次に示します。本例では2～10桁(1～4ﾊ゙ｲﾄ)のﾊ゚ｯｸ
化BCD⇒ﾊ゙ｲﾅﾘ変換が行えます。但し、結果のﾊ゙ｲﾅﾘ値を最大4ﾊ゙ｲﾄとして扱っているため、10桁のﾊ゚ｯｸ化BCDの場
合の最大値は4294967295に制限されます。結果のﾊ゙ｲﾅﾘ値はX+0からになります。元値のﾊ゚ｯｸ化BCD値はY+0が最
下位ﾊ゙ｲﾄです。

BCD桁数 63 4 85 7 92 10

4 521 3ﾊ゚ｯｸ化BCDﾊ゙ｲﾄ数

ﾊ゙ｲﾅﾘ値ﾊ゙ｲﾄ数(ﾋ゙ｯﾄ長) 3 (24) 4 (30)4 (27)2 (10)1 (7) 3 (20)2 (14) 3 (17) 5 (34)

X� DS� 4+4� ;ﾊ゙ｲﾅﾘ値(変換値)変数/一時保存変数定義
Y� DS� 5� ;ﾊ゚ｯｸ化BCD値(元値)変数定義
CNT� DS� 1� ;ﾊ゚ｯｸ化BCD桁数ｶｳﾝﾀ定義

D2B2� MOVLW� 2-1� ;BCD桁数(=2)-1をWに取得� [2桁ﾊ゚ｯｸ化BCD入口]
� GOTO� DB1� ;ﾊ゙ｲﾅﾘ1ﾊ゙ｲﾄ変換処理へ
;
D2B3� MOVLW� 3-1� ;BCD桁数(=3)-1をWに取得� [3桁ﾊ゚ｯｸ化BCD入口]
� GOTO� DB2� ;ﾊ゙ｲﾅﾘ2ﾊ゙ｲﾄ変換処理へ
;
D2B4� MOVLW� 4-1� ;BCD桁数(=4)-1をWに取得� [4桁ﾊ゚ｯｸ化BCD入口]
� GOTO� DB2� ;ﾊ゙ｲﾅﾘ2ﾊ゙ｲﾄ変換処理へ
;
D2B5� MOVLW� 5-1� ;BCD桁数(=5)-1をWに取得� [5桁ﾊ゚ｯｸ化BCD入口]
� GOTO� DB3� ;ﾊ゙ｲﾅﾘ3ﾊ゙ｲﾄ変換処理へ
;
D2B6� MOVLW� 6-1� ;BCD桁数(=6)-1をWに取得� [6桁ﾊ゚ｯｸ化BCD入口]
� GOTO� DB3� ;ﾊ゙ｲﾅﾘ3ﾊ゙ｲﾄ変換処理へ
;
D2B7� MOVLW� 7-1� ;BCD桁数(=7)-1をWに取得� [7桁ﾊ゚ｯｸ化BCD入口]
� GOTO� DB3� ;ﾊ゙ｲﾅﾘ3ﾊ゙ｲﾄ変換処理へ
;
D2B8� MOVLW� 8-1� ;BCD桁数(=8)-1をWに取得� [8桁ﾊ゚ｯｸ化BCD入口]
� GOTO� DB4� ;ﾊ゙ｲﾅﾘ4ﾊ゙ｲﾄ変換処理へ
;
D2B9� MOVLW� 9-1� ;BCD桁数(=9)-1をWに取得� [9桁ﾊ゚ｯｸ化BCD入口]
� GOTO� DB4� ;ﾊ゙ｲﾅﾘ4ﾊ゙ｲﾄ変換処理へ
;
D2B10� MOVLW� 10-1� ;BCD桁数(=10)-1をWに取得� [10桁ﾊ゚ｯｸ化BCD入口]
;
DB4� CLRF� AWKX+3� ;結果ﾊ゙ｲﾅﾘ変数を初期化
DB3� CLRF� AWKX+2� ;
DB2� CLRF� AWKX+1� ;
DB1� CLRF� AWKX+0� ;
� MOVWF� CNT� ;BCD桁ｶｳﾝﾀを設定
;
D2B1� RRF� CNT,W� ;ﾊ゚ｯｸ化BCD値ﾊ゙ｲﾄ ｵﾌｾｯﾄをWに取得
;
� ADDLW� Y� ;ﾊ゚ｯｸ化BCD対応ﾊ゙ｲﾄ位置を取得
� MOVWF� FSR� ;ﾊ゚ｯｸ化BCD対応ﾊ゙ｲﾄ用間接ﾎ゚ｲﾝﾀ設定
� MOVF� INDF,W� ;対応ﾊ゙ｲﾄ下位桁をWに仮取得
� BTFSC� CNT,0� ;下位桁(=0)でｽｷｯﾌ゚
� SWAPF� INDF,W� ;上位桁で対応ﾊ゙ｲﾄ上位桁をW下位に取得

次頁へ続く

;
� CALL� A1D� ;BCD1桁加算
� CALL� M2B� ;現在のﾊ゙ｲﾅﾘ値を2倍
� MOVF� X+0,W� ;最下位ﾊ゙ｲﾄ×2値をWに取得
� MOVWF� X+4� ;最下位ﾊ゙ｲﾄ×2値を一時保存
� MOVF� X+1,W� ;第2ﾊ゙ｲﾄ×2値をWに取得
� MOVWF� X+5� ;第2ﾊ゙ｲﾄ×2値を一時保存
� MOVF� X+2,W� ;第3ﾊ゙ｲﾄ×2値をWに取得
� MOVWF� X+6� ;第3ﾊ゙ｲﾄ×2値を一時保存
� MOVF� X+3,W� ;第4ﾊ゙ｲﾄ×2値をWに取得
� MOVWF� X+7� ;第4ﾊ゙ｲﾄ×2値を一時保存
� CALL� M2B� ;現在のﾊ゙ｲﾅﾘ値を2倍(×4)
� CALL� M2B� ;現在のﾊ゙ｲﾅﾘ値を2倍(×8)
� MOVF� X+4,W� ;加算値最下位ﾊ゙ｲﾄをWに取得
� ADDWF� X+0,F� ;最下位ﾊ゙ｲﾄ加算
;
� MOVF� X+5,W� ;加算値第2ﾊ゙ｲﾄをWに取得
� BTFSC� C� ;前桁からのｷｬﾘーなしでｽｷｯﾌ゚
� INCFSZ� X+5,W� ;ｷｬﾘーで加数第2ﾊ゙ｲﾄ+1をWに取得,=0でｽｷｯﾌ゚
� ADDWF� X+1,F� ;前桁or第2ﾊ゙ｲﾄ+1のｷｬﾘーなしで第2ﾊ゙ｲﾄ加算
;
� MOVF� X+6,W� ;加算値第3ﾊ゙ｲﾄをWに取得
� BTFSC� C� ;前桁からのｷｬﾘーなしでｽｷｯﾌ゚
� INCFSZ� X+6,W� ;ｷｬﾘーで加数第3ﾊ゙ｲﾄ+1をWに取得,=0でｽｷｯﾌ゚
� ADDWF� X+2,F� ;前桁or第3ﾊ゙ｲﾄ+1のｷｬﾘーなしで第3ﾊ゙ｲﾄ加算
;
� MOVF� X+7,W� ;加算値第4ﾊ゙ｲﾄをWに取得
� BTFSC� C� ;前桁からのｷｬﾘーなしでｽｷｯﾌ゚
� INCFSZ� X+7,W� ;ｷｬﾘーで加数第4ﾊ゙ｲﾄ+1をWに取得,=0でｽｷｯﾌ゚
� ADDWF� X+3,F� ;前桁or第4ﾊ゙ｲﾄ+1のｷｬﾘーなしで第4ﾊ゙ｲﾄ加算

� DECFSZ� CNT,F� ;BCD桁ｶｳﾝﾀを減数,=0でｽｷｯﾌ゚
� GOTO� D2B1� ;BCD桁数-1分継続
;
� MOVF� Y+0,W� ;BCD最下位ﾊ゙ｲﾄをWに取得

;� ( BCD1桁加算 )

A1D� ANDLW� 0FH� ;下位桁(BCDの1桁)のみ有効
� ADDWF� X+0,F� ;結果ﾊ゙ｲﾅﾘの最下位ﾊ゙ｲﾄに加算
� MOVLW� 1� ;Wに1を設定
� BTFSC� C� ;ｷｬﾘーなしでｽｷｯﾌ゚
� ADDWF� X+1,F� ;ｷｬﾘーありで第2ﾊ゙ｲﾄに+1
� BTFSC� C� ;ｷｬﾘーなしでｽｷｯﾌ゚
� ADDWF� X+2,F� ;ｷｬﾘーありで第3ﾊ゙ｲﾄに+1
� BTFSC� C� ;ｷｬﾘーなしでｽｷｯﾌ゚
� ADDWF� X+3,F� ;ｷｬﾘーありで第4ﾊ゙ｲﾄに+1
� RETURN� � ;呼出元へ復帰

;� ( ﾊ゙ｲﾅﾘ値2倍 )

M2B� BCF� C� ;ｷｬﾘー ﾌﾗｸ゙を解除
� RLF� X+0,F� ;結果ﾊ゙ｲﾅﾘの最下位ﾊ゙ｲﾄ×2
� RLF� X+1,F� ;第2ﾊ゙ｲﾄ×2
� RLF� X+2,F� ;第3ﾊ゙ｲﾄ×2
� RLF� X+3,F� ;第4ﾊ゙ｲﾄ×2
� RETURN� � ;呼出元へ復帰

■ ﾊ゚ｯｸ化BCD加算

中位PICにはﾊ゚ｯｸ化BCD加算命令がありませんので8ﾋ゙ｯﾄ(1ﾊ゙ｲﾄ)長のﾊ゙ｲﾅﾘ加算命令ADDWF命令と桁ｷｬﾘー ﾌﾗｸ゙を
利用して作成します。1ﾊ゙ｲﾄ長のﾊ゚ｯｸ化BCD変数XとWﾚｼ゙ｽﾀをﾊ゚ｯｸ化BCD加算する例(X+W=X)を以下に示します。

X� DS� 1� ;1ﾊ゙ｲﾄ長ﾊ゚ｯｸ化BCD変数X定義
FLG� DS� 1� ;ﾊ゚ｯｸ化BCDｷｬﾘー ﾌﾗｸ゙変数定義

PADD� ADDWF� X,F� ;ﾊ゙ｲﾅﾘ加算
� MOVF� STATUS,W� ;ﾌﾗｸ゙をWに取得
� MOVWF� FLG� ;ﾌﾗｸ゙を保存
;
� MOVLW� 16-10� ;加算で桁ｷｬﾘーとなる値をWに取得
� ADDWF� X,F� ;下位桁9以下か検査(下位桁補整)
� BTFSC� C� ;下位桁補整でのﾊ゙ｲﾄ ｷｬﾘーなしでｽｷｯﾌ゚
� BSF� FLG,0� ;下位桁補整でのﾊ゙ｲﾄ ｷｬﾘーで結果ｷｬﾘー=1
;
� BTFSC� FLG,1� ;以前のﾊ゙ｲﾅﾘ加算での桁ｷｬﾘーなしでｽｷｯﾌ゚
� GOTO� PADD1� ;ﾊ゙ｲﾅﾘ加算での桁ｷｬﾘーで分岐
;
� BTFSS� DC� ;下位桁10以上でｽｷｯﾌ゚
� SUBWF� X,F� ;下位桁9以下で元値復帰
;
PADD1� MOVLW� (16-10)*16�;加算でｷｬﾘーとなる値をWに取得
� ADDWF� X,F� ;上位桁9以下か検査
� BTFSC� C� ;上位桁9以下でｽｷｯﾌ゚
� BSF� FLG,0� ;上位桁10以上で結果ｷｬﾘー ﾌﾗｸ゙を設定
;
� BTFSS� FLG,0� ;最終的な上位桁からのｷｬﾘーありでｽｷｯﾌ゚
� SUBWF� X,F� ;ｷｬﾘーなしで元値復帰
;

上記形式の連続で多倍長のﾊ゚ｯｸ化BCD加算が行えますが、命令語数的に不利です。通常、多倍長のﾊ゚ｯｸ化BCD
加算はﾙー ﾌ゚処理で記述します。ﾙー ﾌ゚処理形式では任意のﾊ゙ｲﾄ数のﾊ゚ｯｸ化BCDが扱え、実行速度は概ねﾊ゙ｲﾄ数に
比例します。

■ ﾙー ﾌ゚処理によるﾊ゚ｯｸ化BCD加算

ﾊ゚ｯｸ化BCD演算は桁当りの命令数が多いので、ﾙー ﾌ゚処理による演算が一般的です。次の例は任意のﾊ゙ｲﾄ数のﾊ゚ｯｸ
化BCD加算ﾙー ﾁﾝです。このﾙー ﾁﾝの呼び出し時にはWﾚｼ゙ｽﾀにﾊ゙ｲﾄ数を設定する必要があります。

X� DS� 1～� ;ﾊ゚ｯｸ化BCD変数X定義
Y� DS� 1～� ;ﾊ゚ｯｸ化BCD変数Y定義
CNT� DS� 1� ;ﾊ゙ｲﾄ ｶｳﾝﾀ定義
TMP� � � ;一時保存ﾃ゙ ﾀー(FLGと領域を共有)
FLG� DS� 1� ;BCD最終桁からのｷｬﾘー ﾌﾗｸ゙変数定義

PADD� MOVWF� CNT� ;ﾊ゙ｲﾄ ｶｳﾝﾀを設定
� MOVLW� X� ;変数X基準ﾎ゚ｲﾝﾀをWに取得
� MOVWF� FSR� ;変数X基準位置を間接ﾎ゚ｲﾝﾀに設定
� BCF� FLG,0� ;結果ｷｬﾘー ﾌﾗｸ゙を初期化
PADD1� MOVLW� Y-X� ;変数YｵﾌｾｯﾄをWに取得
� ADDWF� FSR,F� ;対応変数Y位置を間接ﾎ゚ｲﾝﾀに設定
� MOVF� INDF,W� ;加算値をWに取得
� BTFSC� FLG,0� ;前回桁からのｷｬﾘーなしでｽｷｯﾌ゚
� ADDLW� 1� ;ｷｬﾘーありでBCD下位桁+1
;
� MOVWF� TMP� ;加算値を一時保存
� MOVLW� Y-X� ;対応変数XｵﾌｾｯﾄをWに取得
� SUBWF� FSR,F� ;対応変数X位置を間接ﾎ゚ｲﾝﾀに設定
� MOVF� TMP,W� ;加算値をWに復帰
� ADDWF� INDF,F� ;ﾊ゙ｲﾅﾘ加算
� MOVF� STATUS,W� ;ﾌﾗｸ゙をWに取得
� MOVWF� FLG� ;ﾌﾗｸ゙を保存
;
� MOVLW� 16-10� ;加算で桁ｷｬﾘーとなる値をWに取得
� ADDWF� INDF,F� ;下位桁9以下か検査(下位桁補整)
� BTFSC� C� ;下位桁補整でのﾊ゙ｲﾄ ｷｬﾘー無しでｽｷｯﾌ゚
� BSF� FLG,0� ;下位桁補整でのﾊ゙ｲﾄ ｷｬﾘーで結果ｷｬﾘー=1
;
� BTFSC� FLG,1� ;以前のﾊ゙ｲﾅﾘ加算での桁ｷｬﾘー検査
� GOTO� PADD2� ;ﾊ゙ｲﾅﾘ加算での桁ｷｬﾘーで分岐
;
� BTFSS� DC� ;下位桁10以上9でｽｷｯﾌ゚
� SUBWF� INDF,F� ;下位桁9以下で元値復帰
;
PADD2� MOVLW� (16-10)*16�;加算でｷｬﾘーとなる値をWに取得
� ADDWF� INDF,F� ;上位桁9以下か検査
� BTFSC� C� ;上位桁9以下でｽｷｯﾌ゚
� BSF� FLG,0� ;上位桁10以上で結果ｷｬﾘー ﾌﾗｸ゙を設定
;
� BTFSS� FLG,0� ;最終的な上位桁からのｷｬﾘーありでｽｷｯﾌ゚
� SUBWF� INDF,F� ;ｷｬﾘーなしで元値復帰
;
� INCF� FSR,F� ;間接ﾎ゚ｲﾝﾀを進行
� DECFSZ� CNT,F� ;ﾊ゙ｲﾄ ｶｳﾝﾀを減数,=0でｽｷｯﾌ゚
� GOTO� PADD1� ;ﾊ゙ｲﾄ数分継続
;
� RETURN� � ;呼出元へ復帰

変数定義は必要とする最大ﾊ゙ｲﾄ数を定義します。

次にﾊ゚ｯｸ化BCDの各ﾊ゙ｲﾄ数に対する実行ｻｲｸﾙ数の関係を示します。

■ ﾊ゚ｯｸ化BCD長と速度の関係

BCDﾊ゙ｲﾄ数 命令語数 実行ｻｲｸﾙ数BCD桁数 BCDﾊ゙ｲﾄ数 命令語数 実行ｻｲｸﾙ数BCD桁数

1,2 34 ～ 351 11,12 179 ～ 1856

3,4 63 ～ 652 13,14 208 ～ 2157

5,6 35 92 ～ 953 15,16 237 ～ 2458 35

7,8 121 ～ 1254 17,18 266 ～ 2759

9,10 150 ～ 1555 19,20 295 ～ 30510

■ ﾊ゚ｯｸ化BCD減算

中位PICにはﾊ゚ｯｸ化BCD減算命令がありませんので、8ﾋ゙ｯﾄ(1ﾊ゙ｲﾄ)長のﾊ゙ｲﾅﾘ減算命令SUBWFと桁ﾎ゙ﾛー (ｷｬﾘー) 
ﾌﾗｸ゙を利用して作成します。1ﾊ゙ｲﾄ長のﾊ゚ｯｸ化BCD変数XからWﾚｼ゙ｽﾀをﾊ゚ｯｸ化BCD減算し、結果を変数Xに格納す
る例(X-W=X)を以下に示します。

X� DS� 1� ;1ﾊ゙ｲﾄ長ﾊ゚ｯｸ化BCD変数X定義
FLG� DS� 1� ;ﾊ゚ｯｸ化BCDﾎ゙ﾛー ﾌﾗｸ゙変数定義

PSUB� SUBWF� X,F� ;ﾊ゙ｲﾅﾘ減算
� BSF� FLG,0� ;結果ﾎ゙ﾛー ﾌﾗｸ゙を初期化
� RLF� FLG,F� ;ﾎ゙ﾛー  ﾌﾗｸ゙を保存
;
� MOVF� X,W� ;ﾊ゙ｲﾅﾘ減算結果をWに取得
� BTFSC� DC� ;ﾊ゙ｲﾅﾘ減算の桁ﾎ゙ﾛーでｽｷｯﾌ゚
� SUBLW� 9� ;下位桁9以下か検査
;
� MOVLW� 16-10� ;下位桁補整値(=6)をWに取得
� BTFSS� DC� ;9以下(桁ﾎ゙ﾛー無し)でｽｷｯﾌ゚
� SUBWF� X,F� ;下位桁9以下に補整
;
� RRF� FLG,F� ;ﾊ゙ｲﾅﾘ減算でのﾎ゙ﾛー ﾌﾗｸ゙を復帰
� MOVF� X,W� ;下位桁補整後ﾃ゙ ﾀーをWに取得
� BTFSC� C� ;ﾊ゙ｲﾅﾘ減算でのﾎ゙ﾛーでｽｷｯﾌ゚
� SUBLW� 9*16� ;上位桁9以下か検査
� BTFSC� C� ;10以上(ﾎ゙ﾛー)でｽｷｯﾌ゚
� GOTO� PSUB1� ;9以下で分岐
;
� MOVLW� (16-10)*16�;上位桁補整値(=60H)をWに取得
� SUBWF� X,F� ;上位桁9以下に補整
� BCF� FLG,0� ;結果ﾎ゙ﾛー ﾌﾗｸ゙を設定
;
PSUB1
;

上記形式の連続で多倍長のﾊ゚ｯｸ化BCD減算が行えますが、命令語数的に不利です。通常、多倍長のﾊ゚ｯｸ化BCD
減算はﾙー ﾌ゚処理で記述します。ﾙー ﾌ゚処理形式では任意のﾊ゙ｲﾄ数のﾊ゚ｯｸ化BCDが扱え、実行速度は概ねﾊ゙ｲﾄ数に
比例します。

■ ﾙー ﾌ゚処理によるﾊ゚ｯｸ化BCD減算

ﾊ゚ｯｸ化BCD演算は桁当りの命令数が多いのでﾙー ﾌ゚処理による演算が一般的です。次の例は任意のﾊ゙ｲﾄ数のﾊ゚ｯｸ
化BCD減算ﾙー ﾁﾝです。このﾙー ﾁﾝの呼び出し時にはWﾚｼ゙ｽﾀにﾊ゙ｲﾄ数を設定する必要があります。

X� DS� 1～� ;ﾊ゚ｯｸ化BCD変数X定義
Y� DS� 1～� ;ﾊ゚ｯｸ化BCD変数Y定義
CNT� DS� 1� ;ﾊ゙ｲﾄ ｶｳﾝﾀ定義
TMP� � � ;一時保存ﾃ゙ ﾀー(FLGと領域を共有)
FLG� DS� 1� ;BCD最終桁からのﾎ゙ﾛー ﾌﾗｸ゙変数定義

PSUB� MOVWF� CNT� ;ﾊ゙ｲﾄ ｶｳﾝﾀを設定
� MOVLW� X� ;変数X基準ﾎ゚ｲﾝﾀをWに取得
� MOVWF� FSR� ;変数X基準位置を間接ﾎ゚ｲﾝﾀに設定
� BSF� FLG,0� ;結果ﾎ゙ﾛー ﾌﾗｸ゙を初期化
PSUB1� MOVLW� Y-X� ;変数間ｵﾌｾｯﾄをWに取得
� ADDWF� FSR,F� ;対応変数Y位置を間接ﾎ゚ｲﾝﾀに設定
� MOVF� INDF,W� ;減算値をWに取得
� BTFSS� FLG,0� ;前回桁からのﾎ゙ﾛーなしでｽｷｯﾌ゚
� ADDLW� 1� ;ﾎ゙ﾛー ありで減算値BCD下位桁+1
;
� MOVWF� TMP� ;減算値を一時保存
� MOVLW� Y-X� ;変数間ｵﾌｾｯﾄをWに取得
� SUBWF� FSR,F� ;対応変数X位置を間接ﾎ゚ｲﾝﾀに設定
� MOVF� TMP,W� ;減算値をWに復帰
� SUBWF� INDF,F� ;ﾊ゙ｲﾅﾘ減算
� BSF� FLG,0� ;結果ﾎ゙ﾛー ﾌﾗｸ゙を初期化
� RLF� FLG,F� ;ﾎ゙ﾛー  ﾌﾗｸ゙を保存
;
� MOVF� INDF,W� ;ﾊ゙ｲﾅﾘ減算結果をWに取得
� BTFSC� DC� ;ﾊ゙ｲﾅﾘ減算の桁ﾎ゙ﾛーでｽｷｯﾌ゚
� SUBLW� 9� ;下位桁9以下か検査
;
� MOVLW� 16-10� ;下位桁補整値(=6)をWに取得
� BTFSS� DC� ;9以下(桁ﾎ゙ﾛーなし)でｽｷｯﾌ゚
� SUBWF� INDF,F� ;下位桁9以下に補整
;
� RRF� FLG,F� ;ﾊ゙ｲﾅﾘ減算でのﾎ゙ﾛー ﾌﾗｸ゙を復帰
� MOVF� INDF,W� ;下位桁補整後ﾃ゙ ﾀーをWに取得
� BTFSC� C� ;ﾊ゙ｲﾅﾘ減算でのﾎ゙ﾛーでｽｷｯﾌ゚
� SUBLW� 9*16� ;上位桁9以下か検査
� BTFSC� C� ;10以上(ﾎ゙ﾛー)でｽｷｯﾌ゚
� GOTO� PSUB2� ;9以下で分岐
;
� MOVLW� (16-10)*16�;上位桁補整値(=60H)をWに取得
� SUBWF� INDF,F� ;上位桁9以下に補整
� BCF� FLG,0� ;結果ﾎ゙ﾛー ﾌﾗｸ゙を設定
;
PSUB2� INCF� FSR,F� ;間接ﾎ゚ｲﾝﾀを進行
� DECFSZ� CNT,F� ;ﾊ゙ｲﾄ ｶｳﾝﾀを減数,=0でｽｷｯﾌ゚
� GOTO� PSUB1� ;ﾊ゙ｲﾄ数分継続
;
� RETURN� � ;呼出元へ復帰

変数定義は必要とする最大ﾊ゙ｲﾄ数を定義します。

次にﾊ゚ｯｸ化BCDの各ﾊ゙ｲﾄ数に対する実行ｻｲｸﾙ数の関係を示します。

■ ﾊ゚ｯｸ化BCD長と速度の関係

BCDﾊ゙ｲﾄ数 命令語数 実行ｻｲｸﾙ数BCD桁数 BCDﾊ゙ｲﾄ数 命令語数 実行ｻｲｸﾙ数BCD桁数

1,2 34 ～ 361 11,12 179 ～ 1916

3,4 63 ～ 672 13,14 208 ～ 2227

5,6 35 92 ～ 983 15,16 237 ～ 2538 35

7,8 121 ～ 1294 17,18 266 ～ 2849

9,10 150 ～ 1605 19,20 295 ～ 31510

■ TMR0の使用法

TMR0の使用で注意しなければいけないのは設定値です。基本的なことですが、TMR0は上昇ｶｳﾝﾀで0FFHからの
ｵー ﾊ゙ ﾌーﾛー で割り込みが発生します。このため、設定値は256-目的の計数値になります。実際のﾌ゚ﾛｸ゙ﾗﾑ上では単に
目的の計数値に負符号を追加するだけです。例えば目的の計数値が16の場合、設定値は256-16=240ですが、-16と
の記述は内部で0FFFFFFE0H(32ﾋ゙ｯﾄ表現)になり、その下位8ﾋ゙ｯﾄが採用され、結果は0E0Hで10進数の240になりま
す。

実際に注意が必要なのは同期化のための停止時間と割り込み応答時間です。これらによって結果的にﾀｲﾏ間隔が長
くなる問題が発生します。特に前置分周器を有効(2分周以上)にする場合は大変複雑になります。

次に前置分周器なしでの上記問題点を示すﾀｲﾐﾝｸ゙を示します。

■ 前置分周器なしでの同期化遅延と割り込み応答遅延のﾀｲﾐﾝｸ゙

2310

命令 命令 命令 命令
MOVWF
TMR0 命令 命令 命令

0FEH 0FFH 00H 01H 02H n n n n+1

2310 2310 2310 2310 2310 2310 2310 210

命令

FOSC

命令実行

TMR0値

TOIF

割り込み応答遅延 TMR0同期化遅延割り込み処理

この場合、TMR0同期化遅延の2ｻｲｸﾙと割り込み応答遅延の3ｻｲｸﾙ分を目的計数値に対して補正すれば正確な
ﾀｲﾏ周期が得られます。両方共に遅延要素ですので、実際の設定値は上記での設定値+5になります。上記例の16の
場合は5-16のようになります。

現実には、この例の割り込み処理のようにいきなりTMR0書き込み命令となることはなく、ﾚｼ゙ｽﾀ保存やﾀｲﾏ設定値の
取得命令などの後になります。その場合、それらの命令分の遅延が割り込み応答遅延に加算されます。これらを含め
て最終的な設定値を求めます。

しかし、割り込み処理ﾙー ﾁﾝ先頭からTMR0書き込み命令間の命令実行ｻｲｸﾙ数に応じて設定値を都度調整すること
になり非常に不便です。また、2ｻｲｸﾙ命令実行時に割り込みが発生した場合は割り込み応答遅延が+1されるので、こ
の方法では常に正確なﾀｲﾏ周期が得られません。これらを解決するにはTMR0に書き込む(MOVWF TMR0)のではな
く、TMR0に加算する(ADDWF TMR0,F)方法にします。TMR0ｵー ﾊ゙ ﾌーﾛー 後、TMR0は継続して00Hからの上昇計数を
行っていますので、これはｵー ﾊ゙ ﾌーﾛー 後の経過ｻｲｸﾙ数を表します。このため、補正値は常にTMR0同期化遅延の
2ｻｲｸﾙ分だけになります。

この方法でも前置分周器を使用する場合は更に話が複雑になります。上記2項目の遅延自体は変わりませんが、
TMR0書き込み命令とTMR0計数ｸﾛｯｸ(前置分周器出力)との同期化も行わなければなりません。TMR0書き込みによ
って前置分周器がﾘｾｯﾄされますので、TMR0書き込み命令はTMR0計数時と一致させる必要があります。ところが
TMR0同期化遅延は常に2ｻｲｸﾙですので、実際にはTMR0計数時の2ｻｲｸﾙ前で同期化を行わなければなりません。
前置分周器が2分周の場合はTMR0計数時と一致させ、設定値を+1することでも行なえます。

前置分周器が2分周で問題(1ｻｲｸﾙ誤差)となるﾀｲﾐﾝｸ゙例を次に示します。

■ 前置分周器が2分周での1ｻｲｸﾙ誤差発生ﾀｲﾐﾝｸ゙

2310

命令 命令 命令 命令

前置分周器が2分周時のTMR0書き込み命令とTMR0計数ｸﾛｯｸ(前置分周器出力)との同期化ﾌ゚ﾛｸ゙ﾗﾑ例を次に示しま
す。

SYNC� MOVF� TMR0,W� ;現在のTMR0値をWに取得
� SUBWF� TMR0,W� ;このｻｲｸﾙで計数か検査
� BTFSC� Z� ;このｻｲｸﾙが計数でｽｷｯﾌ゚�
� GOTO� SYNC1� ;このｻｲｸﾙが計数で分岐
;
SYNC1� MOVLW� 1-16� ;(設定値-TMR0同期化遅延補正)値をWに取得
� ADDWF� TMR0,F� ;TMR0再設定/補正 (このｻｲｸﾙが計数ｻｲｸﾙ)

この例ではTMR0同期化遅延の2ｻｲｸﾙ分が2分周の1ｶｳﾝﾄに相当するので、設定値を-1することで補正しています。4
分周以上では上記例のような簡単な記述では行なえず、かなり複雑になります。次に4分周時の例を示します。

SYNC� MOVF� TMR0,W� ;現在のTMR0値をWに取得
� SUBWF� TMR0,W� ;このｻｲｸﾙで計数か検査
� BTFSS� Z� ;直前の命令ｻｲｸﾙが計数でなければｽｷｯﾌ゚�
� GOTO� SYNC1� ;2つ前の命令ｻｲｸﾙが計数で分岐
;
� NOP� � ;SUBWF命令間が4+1ｻｲｸﾙとなるための補正
� MOVF� TMR0,W� ;現在のTMR0値をWに取得
� SUBWF� TMR0,W� ;このｻｲｸﾙで計数か検査
� BTFSS� Z� ;直前の命令ｻｲｸﾙが計数でなければｽｷｯﾌ゚�
� GOTO� SYNC1� ;2つ前の命令ｻｲｸﾙが計数で分岐
;
� NOP� � ;SUBWF命令間が4+1ｻｲｸﾙとなるための補正
� MOVF� TMR0,W� ;現在のTMR0値をWに取得
� SUBWF� TMR0,W� ;このｻｲｸﾙで計数か検査
� BTFSS� Z� ;直前の命令ｻｲｸﾙが計数でなければｽｷｯﾌ゚�
� GOTO� SYNC1� ;2つ前の命令ｻｲｸﾙが計数で分岐
;
� NOP� � ;最終同期位置用補正
� NOP� � ;(直前のGOTO SYNC1より+1ｻｲｸﾙとなる補正)
;
SYNC1� NOP� � ;一致時のSUBWF命令からADDWF命令間ｻｲｸﾙ数補正
� MOVLW� -16� ;(4の倍数-TMR0同期化遅延の2ｻｲｸﾙ)正の最小値
� ADDWF� TMR0,F� ;TMR0再設定/補正 (このｻｲｸﾙが計数ｻｲｸﾙ)

この例では計数ｻｲｸﾙを検査するSUBWF命令間が4+1命令ｻｲｸﾙとなるようにしています。この計数ｻｲｸﾙ検査を3回
行うことで計数ｻｲｸﾙ位置を得ています。3回目で一致しなかった場合は次のｻｲｸﾙが該当位置であることが分かって
いますので、単にNOP命令でｻｲｸﾙ数調整を行っています。この例のMOVLWの取得値は負符号を付加した目的の
計数値そのものです。TMR0同期化遅延はﾌ゚ﾛｸ゙ﾗﾑ実行ｻｲｸﾙで補正しています。

これ以上の分周値でも上記例と同様な手法で対応可能ですが、命令語数と実行時間からあまり実用的とは言えませ
ん。むしろﾌ゚ﾛｸ゙ﾗﾑによるｶｳﾝﾀを併用し、TMR0自体はできるだけ前置分周器なしで使用する方が得策です。

何れの方法を採っても、割り込み発生時に実行中の命令の実行ｻｲｸﾙが1の場合と2の場合でTMR0と割り込み処理
ﾙー ﾁﾝ開始ﾀｲﾐﾝｸ゙は1ｻｲｸﾙの誤差を生じますので、ﾌ゚ﾛｸ゙ﾗﾑによるﾎ゚ ﾄー出力などでのTMR0との同期化を完全/正確
に行うことは不可能です。

■ 入出力の拡張

入出力の拡張方法には多くの方法がありますが、ここでは一般的な同期ｼﾘｱﾙ接続による方法を示します。ﾊ゚ﾗﾚﾙ接
続の方が速度的に有利ですが、内蔵ﾎ゚ー ﾄを大量に消費しますので得策ではありません。ｼﾘｱﾙ接続で8ﾋ゙ｯﾄの入出力
処理を行う場合のﾌ゚ﾛｸ゙ﾗﾑ実行時間は10MHzｸﾛｯｸで約50μs程度になります。

ｼﾘｱﾙ接続の入力ﾃ゙ﾊ゙ｲｽには74HC165,166などが使用できます。出力には74HC595が最適です。また、特定ﾒー ｶー で
は74HC595ﾋ゚ﾝ互換や機能相当で出力形態がﾊ゙ｲﾎ゚ ﾗーTrやFET駆動の大電流ﾄ゙ﾗｲﾊ゙となっているﾃ゙ﾊ゙ｲｽもあります
ので、入手が可能ならば目的に応じて選択します。

次に74HC165と74HC595を使用する場合の例を示します。

上記例は入出力共に16ﾋ゙ｯﾄの拡張例です。この方法ではIC1個が8ﾋ゙ｯﾄ分ですので、8ﾋ゙ｯﾄ単位で任意の拡張ができ
ます。3個目以降は2個目同様に縦列接続で結線します。

注意しなければならないのは順番で、図上のSIは左端のﾃ゙ﾊ゙ｲｽからの入力が先に入力され、SOは左端のﾃ゙ﾊ゙ｲｽが
最後の出力となることです。

上図では入力ﾎ゚ ﾄーのSIをQHに接続しています。これは入力が負論理入力の場合の接続方法で、正論理の場合は
QH(9番ﾋ゚ﾝ)に接続します。このように入力の論理により論理を選択できる点で74HC165は74HC166より有利です。

この出力ﾎ゚ ﾄーの出力が出力ﾎ゚ ﾄー初期化まで不定でもよい場合はOEをGNDに接続できます。また、出力ﾎ゚ ﾄーの初期
出力値がLならば、SCLRをﾘｾｯﾄ信号に接続することで、OEをGNDに接続できます。

PIC内蔵ﾎ゚ー ﾄの使用を少しでも減らしたい場合にはSIとSOを接続し、1ﾋ゙ｯﾄで処理することもできます。その場合のSIは
10kΩ程度の抵抗を介して接続します。ﾌ゚ﾛｸ゙ﾗﾑでは方向ﾚｼ゙ｽﾀで内蔵ﾎ゚ ﾄーの該当ﾋ゙ｯﾄの方向を切り替えながら入出
力を行います。この方法では1つのｼﾌﾄ ｸﾛｯｸ内で入出力を同時に処理できず、入力と出力を個別に処理しなければ
ならないので、入出力処理時間が入出力同時処理に比べて倍近くかかります。

■ 拡張入出力制御 ﾀｲﾐﾝｸ゙図

① 74HC165はS/L=L(実際には立ち上がり)で165のA～H入力をﾗｯﾁします。②のﾀｲﾐﾝｸ゙で代用可能ですが、この場
合は前回の時点での入力状態になります。最新の状態を入力するためには①のﾀｲﾐﾝｸ゙で行います。同時に
74HC595でもｼﾌﾄ ﾚｼ゙ｽﾀ内のﾃ゙ ﾀーがﾗｯﾁに取り込まれますが、ｼﾌﾄ ﾚｼ゙ｽﾀの状態が変化していませんので、これ
によって595の出力は変化しません。

② 74HC595のRCKの立ち上がりで595のｼﾌﾄ ﾚｼ゙ｽﾀ値をﾗｯﾁ ﾚｼ゙ｽﾀにﾗｯﾁします。この結果、595の各出力が同時に
変化します。同時に74HC165でもA～H入力のﾗｯﾁが行われます。①を行わない場合、これが165入力ﾗｯﾁ動作に
なります。

74HC165のCLOCKと74HC595のSCKはｼﾌﾄ ﾚｼ゙ｽﾀに対するｼﾌﾄ ｸﾛｯｸで、共に立ち上がりでｼﾌﾄ動作を行います。

■ 拡張入出力初期化ﾀｲﾐﾝｸ゙図

①　ﾘｾｯﾄ後のﾌ゚ﾛｸ゙ﾗﾑ開始で通常のようにﾎ゚ ﾄー初期化を行います。その後に接続した拡張入出力ﾋ゙ｯﾄの入出力処理
を行います(上図例は8ﾋ゙ｯﾄ分)。この時の出力ﾃ゙ ﾀーが74HC595の初期値になります。また、入力ﾃ゙ ﾀーも有効になり
ます。

②　74HC595のRCKの立ち上がりで595の内部ﾚｼ゙ｽﾀに指定値がﾗｯﾁされますが、この時点での595出力端子は未だ
Hi-Z状態です。この後に595のOEをLにすることで595出力端子が有効になります。

■ INT0,TMR0利用のRS-232C

PIC(16F84)にはﾊー ﾄ゙ｳｪｱによるｼﾘｱﾙ ｲﾝﾀー ﾌｪーｽ機能がありませんので、外部割込みとﾀｲﾏを利用し、ｿﾌﾄｳｪｱで実
現しますが、同時送受信は不可で交互双方向通信になります。

送信の場合、ﾋ゙ｯﾄ時間毎に出力信号(TxD)を設定すればよいので、ﾀｲﾏの利用だけで実現できます。このﾀｲﾏの設定
値は1ﾋ゙ｯﾄ時間です。

受信の場合は1ﾊ゙ｲﾄ受信の起動となるｽﾀーﾄ ﾋ゙ｯﾄ開始の検出が必要です。具体的にはｽﾀーﾄ ﾋ゙ｯﾄの立ち上がりを外部
割込み(INT0)で検出し、ﾋ゙ｯﾄ値採取用ﾀｲﾏを起動します。この立ち上がりは回線上での場合で、受信側にｲﾝﾊ゙ ﾀーが
存在する場合の入力信号(RxD)では立ち下りになります。

INT0割り込み

■ 送受信共にｲﾝﾊ゙ ﾀーを使用した場合の基本動作

t1: 1ﾋ゙ｯﾄ時間 t2: 1/2ﾋ゙ｯﾄ時間 ST: ｽﾀーﾄ ﾋ゙ｯﾄSP: ｽﾄｯﾌ゚ ﾋ゙ｯﾄ: ﾋ゙ｯﾄ値採取 : 出力

① INT0割り込みでTMR0の計数値を1/2ﾋ゙ｯﾄ時間に設定します。設定後、必ずTMR0割り込み要求ﾌﾗｸ゙(T0IF)をｸﾘｱ
します。以降のTMR0割り込みでは計数値を1ﾋ゙ｯﾄ時間に設定します。

② ｽﾀーﾄ ﾋ゙ｯﾄ開始後にも殆どの場合、INT0割り込みが発生します。これらのINT0割り込みは1ﾊ゙ｲﾄ受信完了までINT0
割り込みを禁止するか、または無視する必要があります。

TMR0の計数値は基本的に1/2または1ﾋ゙ｯﾄ時間ですが、実際には割り込み応答時間を考慮する必要があります。

送信時は各ﾋ゙ｯﾄ出力処理間の最大誤差が実際の出力信号での誤差となりますが、通常、この差はﾌ゚ﾛｸ゙ﾗﾑ数ｽﾃｯﾌ゚
分程度ですので、転送ﾚーﾄが低い場合は無視できる値です。

受信時の各ﾋ゙ｯﾄ受信処理も同様ですが、全体の基準となるINT0割り込みでのTMR0計数値への1/2ﾋ゙ｯﾄ時間設定は
厳密さが要求されます。

①

■ 上記基本動作でのINT0割り込み時詳細

tC: PIC固有割り込み応答時間 tI: INT0応答時間

INT0処理 TMR0処理 その他処理

tI

tS

tS: TMR0採取応答時間

ｽﾀーﾄ ﾋ゙ｯﾄ中央

tC+tS

③

1/2ﾋ゙ｯﾄ時間
tT

tT: TMR0設定応答時間

① INT0入力の有効ｴｯｼ゙から割り込み処理ﾙー ﾁﾝが起動されるまでに(1または2)＋3ｻｲｸﾙ(×4ｸﾛｯｸ)の遅延(tC)があり
ます。

② ｽﾀーﾄ ﾋ゙ｯﾄ検出のTMR0割り込みはtC+tS分の遅延が存在するため、③位置で発生させなければなりません。また、
TMR0計数値設定時までにtC+tI分の遅延もあります。

これらのため、②での設定値は1/2ﾋ゙ｯﾄ時間-(tC+tI)-(tC+tS)になります。同様にTMR0割り込みでの1ﾋ゙ｯﾄ時間も
1ﾋ゙ｯﾄ時間-(tC+tT)にしなければなりません。

その他の割り込み処理がINT0やTMR0と非同期に発生する場合、その他の割り込み処理中にINT0やTMR0割り込み
を検査し、これらの割り込みがある場合にその処理を優先するなどの処置をすることで、tI,tS,tTなどの応答時間を一
定とする工夫が必要です。

同様にﾌ゚ﾛｸ゙ﾗﾑ内で割り込み禁止期間が存在すると、その期間分tCに対する誤差要因となり、全体に影響を及ぼしま
す。原則的に割り込み禁止の使用は不可としてﾌ゚ﾛｸ゙ﾗﾐﾝｸ゙します。

■ 命令一覧表

�ﾆーﾓﾆｯｸ/ｵﾍ゚ﾗﾝﾄ゙� 意味� ｻｲｸﾙ数� 14ﾋ゙ｯﾄ ｵﾍ゚ｺーﾄ゙� 更新ｽﾃーﾀｽ� 注意

� ﾊ゙ｲﾄ単位のﾌｧｲﾙ ﾚｼ゙ｽﾀ命令

  ADDWF� f,d� Add W and f�    1� 00 0111 dfff ffff�   Z,DC,C� 1,2
  ANDWF� f,d� AND W with f�    1� 00 0101 dfff ffff�   Z� 1,2
  CLRF� f� Clear f�    1� 00 0001 1fff ffff�   Z�   2
  CLRW� � Clear W�    1� 00 0001 0xxx xxxx�   Z
  COMF� f,d� Complement f�    1� 00 1001 dfff ffff�   Z� 1,2
  DECF� f,d� Decrement f�    1� 00 0011 dfff ffff�   Z� 1,2
  DECFSZ�f,d� Decrement f, Skip if 0�   1/2� 00 1011 dfff ffff� � 1,2,3
  INCF� f,d� Increment f�    1� 00 1010 dfff ffff�   Z� 1,2
  INCFSZ�f,d� Increment f, Skip if 0�   1/2� 00 1111 dfff ffff� � 1,2,3
  IORWF� f,d� Inclusive OR W with f�    1� 00 0100 dfff ffff�   Z� 1,2
  MOVF� f,d� Move f�    1� 00 1000 dfff ffff�   Z� 1,2
  MOVWF� f� Move W to f�    1� 00 0000 1fff ffff
  NOP� � No Operation�    1� 00 0000 0xx0 0000
  RLF� f,d� Rotate Left f through Carry�    1� 00 1101 dfff ffff�        C� 1,2
  RRF� f,d� Rotate Right f through Carry�    1� 00 1100 dfff ffff�        C� 1,2
  SUBWF� f,d� Subtract W from f�    1� 00 0010 dfff ffff�   Z,DC,C� 1,2
  SWAPF� f,d� Swap nibbles in f�    1� 00 1110 dfff ffff� � 1,2
  XORWF� f,d� Exclusive OR W with f�    1� 00 0110 dfff ffff�   Z� 1,2

� ﾋ゙ｯﾄ単位のﾌｧｲﾙ ﾚｼ゙ｽﾀ命令

  BCF� f,b� Bit Clear f�    1� 01 00bb bfff ffff� � 1,2
  BSF� f,b� Bit Set f�    1� 01 01bb bfff ffff� � 1,2
  BTFSC� f,b� Bit Test f, Skip if Clear�   1/2� 01 10bb bfff ffff� �     3
  BTFSS� f,b� Bit Test f, Skip if Set�   1/2� 01 11bb bfff ffff� �     3

� ﾘﾃﾗﾙ、制御命令

  ADDLW� k� Add literal and W�    1� 11 111x kkkk kkkk�   Z,DC,C
  ANDLW� k� AND literal with W�    1� 11 1001 kkkk kkkk�   Z
  CALL� k� Call subroutine�    2� 10 0kkk kkkk kkkk
  CLRWDT� � Clear Watchdog Timer�    1� 00 0000 0110 0100�   TO,PD
  GOTO� k� Go to address�    2� 10 1kkk kkkk kkkk
  IORLW� k� Inclusive OR literal with W�    1� 11 1000 kkkk kkkk�   Z
  MOVLW� k� Move literal to W�    1� 11 00xx kkkk kkkk
  RETFIE� � Return from interrupt�    2� 00 0000 0000 1001
  RETLW� k� Return with literal in W�    2� 11 01xx kkkk kkkk
  RETURN� � Return from Subroutine�    2� 00 0000 0000 1000
  SLEEP� � Go into standby mode�    1� 00 0000 0110 0011�   TO,PD
  SUBLW� k� Subtract W from literal�    1� 11 110x kkkk kkkk�   Z,DC,C
  XORLW� k� Exclusive OR literal with W�    1� 11 1010 kkkk kkkk�   Z

b

ｼﾝﾎ゙ﾙ

d

f

k

8ﾋ゙ｯﾄ ﾌｧｲﾙ ﾚｼ゙ｽﾀ内のﾋ゙ｯﾄ ｱﾄ゙ﾚｽ

結果格納先指定子; d = 0 (結果はWに格納)、d = 1 (結果はf (ﾌｧｲﾙ ﾚｼ゙ｽﾀ)に格納)

ﾌｧｲﾙ ﾚｼ゙ｽﾀのｱﾄ゙ﾚｽ (00H ～ 7FH)

ﾘﾃﾗﾙ、定数ﾃ゙ ﾀーまたはﾗﾍ゙ﾙ

W 作業ﾚｼ゙ｽﾀ (累積器)

x 無効 (0または1)

意味

注意: 1. I/Oﾚｼ゙ｽﾀに対するRead modify Write(MOVF RB,Fなど)はﾋ゚ﾝの入力ﾚﾍ゙ﾙが使用されます。例えば入力ﾋ゚ﾝのﾃ゙ ﾀー 
ﾗｯﾁが1、そのﾋ゚ﾝが外部ﾃ゙ﾊ゙ｲｽによりLowﾚﾍ゙ﾙとなっている場合、ﾃ゙ ﾀー ﾗｯﾁには0が書き込まれます。

2. 前置分周器がTMR0に割り当てられている場合、これらの命令をTMR0ﾚｼ゙ｽﾀに対して実行すると(TMR0に対して書
き込みを伴う場合)、TMR0部に割り当てられている前置分周器がｸﾘｱされます。

3. ﾌ゚ﾛｸ゙ﾗﾑ ｶｳﾝﾀ(PC)を変更したり、条件付き試験の結果が真になると、命令実行は2ｻｲｸﾙ掛かり、2番目のｻｲｸﾙは
NOPとして実行されます。

本書は中位PIC(16F84)留意点の記録です。

青字の部分はﾘﾝｸとなっています。一般的に赤字の0,1は論理0,1を表します。その他の赤字は重要な部分を表します。

